


Elektronica projecten voor de **ZX SPECTRUM** en **ZX SPECTRUM+**

W.H.M. van Dreumel

met
interface
print



W.H.M. van Dreumel

**Elektronicaprojecten
voor de
ZX SPECTRUM
en ZX SPECTRUM+**



KLUWER TECHNISCHE BOEKEN

Omslag: W. Niessink

ISBN 90 201 1874 9
D/1986/0108/197

© 1986 Kluwer Technische Boeken B.V. - Deventer

1e druk 1986

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotokopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

No part of this book may be reproduced in any form, by print, photoprint, microfilm or any other means without written permission from the publisher.

Ondanks alle aan de samenstelling van de tekst bestede zorg, kan noch de redactie noch de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout, die in deze uitgave zou kunnen voorkomen.

Inhoud

1. Adressen voor de interface.	9
2. Interface-hardware	12
3. Een analoog uitgangssignaal.	17
4. Het interface-bouwproject	19
5. Uitgangsmonitor	22
6. Test	24
7. Joy-stick	26
8. Ezeltje prik	29
9. Teken met de joy-stick.	32
10. Pixel-restanten	36
11. Spokenjacht	38
12. Het meten van de tijd	41
13. Stopwatch	43
14. De tijd grafisch weergegeven	45
15. Optisch detecteren	48
16. Schakelaar met instelbaar triggerniveau	50
17. Sinusgenerator met faseverschuiving	52
18. Dubbelsinusgenerator	55
19. Een karaktervolle lichtpen	58
20. De pen als dirigeerstok.	62
21. Lichtnetschakelaar	64
22. Toneelverlichting met de pen	67
23. Analoge ingang	70
24. De bouw van de A/D-omzetter	76
25. Voltmeter-software.	81
26. Windsnelheidsmeter	83
27. Aquariumregeling	86
28. Programmeerbare diavloeiër	91
29. Uit is in: Een lichteffect voor de disco	98
30. Verlengprint	102
Tot slot.	103

Woord vooraf

Zelden mocht een computer zo'n grote populariteit genieten als de ZX Spectrum. Geen wonder. Achteenveertig kilo computer-power in een klein zwart doosje. De zwarte doos (black box) wordt meestal beschouwd als een onbekend apparaat dat op commando bepaalde functies verricht. Voor de meeste gebruikers is de Spectrum dan ook niet meer dan een zwarte doos. Kennis over het gebeuren in de centrale verwerkingseenheid, de registers, het videogeheugen en wat er zoal meer in een computer zit, is voorbehouden aan een selecte groep echte kenners. Dat willen we graag zo houden. Echter, al lezend en bouwend zult u spelenderwijs toch vertrouwd raken met een deel van het binnenste van de computer. U zult bepaalde handig te gebruiken adressen leren kennen en een indruk krijgen over de manier waarop de computer contact kan houden met de buitenwereld. Het is juist dit contact waarop in dit boek de nadruk ligt.

De universele interfaceprint vormt de basis van dit werk. Op deze print vinden we een digitale ingang, een joy-stick-aansluiting, een digitale uitgang, maar ook een analoge uitgang. Een scala van opsteekprintjes maakt een zeer veelzijdige inzet van de computer in ons dagelijks leven mogelijk.


Het karakter van dit boek brengt met zich mee dat de projecten een enigszins speels karakter hebben. Per slot van rekening zal de lezer er een aantal plezierige uren mee door moeten brengen. Toch ligt er een serieuze gedachte aan dit boek ten grondslag. Elektronicatijdschriften besteden gelukkig veel aandacht aan computer-georiënteerde elektronica-ontwerpen. Vaak echter zijn deze ontwerpen zo complex dat de minder geroutineerde elektronicus er niet goed mee uit de voeten kan. Dit boek probeert de kloof te dichten tussen de elektronicahobby en de computer. De ontwerpen zijn trefzeker na te bouwen met standaardonderdelen. Bovendien zijn ze geïllustreerd met uitgewerkte programma's.

Terwille van de duidelijkheid wijken de afgedrukte programma's iets af van hetgeen op het scherm zichtbaar wordt. De regels zijn anders afgebroken. Spaties zijn met [] aangegeven. Grafische karakters zijn van een kadertje voorzien.

Tot slot: Elektronica is een moeilijk vak. Hoewel niet te verwachten kan het gebeuren dat u met bepaalde problemen komt te zitten ten aanzien van onderdelen of wijzigingen die u zelf wilt aanbrengen. Laat dat dan een stimulans zijn om aansluiting te zoeken bij geestverwanten: in uw buurt of binnen de gebruikersclub. Laat de computer iets voor u betekenen, maar... blijf hem wel de baas.

1 Adressen voor de interface

Op de achterzijde van de Spectrum bevindt zich een aansluitmogelijkheid voor een brede connector in de vorm van een uitstekend deel van de print. Op deze printstrip kunnen we de meeste interne signalen van de computer terugvinden. Afbeelding 1.1 geeft de aansluitingen weer zoals ze gesitueerd zijn als we de Spectrum aan de achterzijde bekijken. Volledigheidshalve is ook het aanzicht van de connector gegeven, die uiteraard het spiegelbeeld van de eerstgenoemde is.



Afb. 1.1 De aansluitingen van de connector.

De lijnen waarop de computer zijn gegevens behandelt, zijn de DATA-lijnen D0 t/m D7. Deze acht lijnen laten zien dat de computer werkt met woorden die uit acht BITS zijn opgebouwd. Omdat gegevens in het geheugen van de computer op bepaalde adressen worden opgeborgen, heeft de Spectrum een aantal ADRES-lijnen waarop de gewenste geheugenlocatie kan worden aangegeven. Er zijn zestien adreslijnen namelijk A0 t/m A15. Van alle andere signalen zijn voor de interface alleen nog IOREQU, RD en WR van belang.

IOREQ is de afkorting voor IN OUT REQUEST en is als het ware een verzoek van de computer om in- en uitgaande lijnen te mogen gebruiken.

RD staat voor READ. Als de computer gegevens binnenhaalt, is dit leessignaal hoog. Het WRITE-signaal WR is hoog als de computer informatie wegschrijft. In tegenstelling tot de ZX81, heeft de Spectrum de beschikking over de BASIC-commando's IN en OUT. De signalen RD en WR kunnen vanuit een programma wor-

den bestuurd met deze IN- en OUT-opdracht waarbij IN lezen en OUT schrijven betekent. Bij de ZX81 wordt hetzelfde bereikt met de opdrachten PEEK en POKE, waarmee informatie van de connector wordt gelezen en naar de connector wordt geschreven. De connector wordt dan als een deel van het geheugen beschouwd. We kiezen in dat geval een geheugenadres dat niet door de ZX81 zelf wordt gebruikt.

Het Spectrum handboek geeft globaal aan wat er met de adreslijnen A0 t/m A15 zoal gebeurt. De eerste vijf lijnen, A0 t/m A4, worden door de computer zelf gebruikt. Deze adreslijnen zijn normaal gesproken hoog. Als één van deze lijnen laag is, onderneemt de computer een bepaalde actie. Over lijn A5 hult het handboek zich in diep stilzwijgen. De lijnen A6 en A7 hebben geen bepaalde functie en Sinclair adviseert dan ook deze lijnen voor interfacegebruik te benutten, mits er zodanige adreswaarden worden gekozen dat de eerste vier lijnen hoog blijven. Omdat de betekenis van lijn A5 niet duidelijk is zullen we hem bij het bepalen van een geschikt interface-adres negeren. Om de opzet eenvoudig te houden, gebruiken we uitsluitend lijn A6 als adreslijn waarmee de interface wordt aangesproken. De resterende lijnen A8 t/m A15 hebben voor ons, en ten dele ook voor de computer, geen bijzondere betekenis.

Voor het vaststellen van een geschikt adres voor de besturing van de interface kijken we nog even naar de betekenis die de adreslijnen hebben.

De informatie op de adreslijnen is binair. Dat wil zeggen dat het lijnnummer tevens de macht voorstelt waartoe het getal twee dient te worden verheven. De logica is positief zodat een hoog niveau betekent dat de desbetreffende waarde van de lijn als bijdrage in het adres wordt meegerekend en een laag niveau aangeeft dat het desbetreffende kanaal de waarde nul heeft.

Zijn bijvoorbeeld de lijnen A0 t/m A4 hoog (zoals Sinclair voorschrijft) en alle andere lijnen laag, dan is het adres:

Adreslijn	0	1	2	3	4	5	6	7	8	15
Binaire waarde	2^0	$+2^1$	$+2^2$	$+2^3$	$+2^4$	$+0$	$+0$	$+0$	$+0$	$+0$

Vertaald naar het decimale stelsel vinden we dan:

$$\text{Decimale waarde} = 1 + 2 + 4 + 8 + 16 + 0 + 0 + 0 + 0 + \dots + 0 = 31$$

Als we de voor de interface gekozen adreslijn A6 hoog willen hebben, dan dient het adresgetal er als volgt uit te zien:

$$1 + 2 + 4 + 8 + 16 + 0 + 64 + 0 + 0 + \dots + 0 = 95$$

Op zich is er geen bezwaar tegen een ander adres te kiezen zolang de lijnen A0,1,2,3,4 en 6 maar hoog zijn en een aantal gereserveerde adressen wordt vermeden. In hoofdstuk 23 'IN and OUT' van het Engelstalige gebruikershandboek zijn de gereserveerde adressen vermeld.

Het is niet erg aantrekkelijk om hoge adreslijnen te kiezen. Bij de hogere lijnnummers worden de getallen nogal groot en zijn dan moeilijker te onthouden en lastiger in te typen.

Met het vorenstaande zou dan het door ons te gebruiken interface-adres zijn vastgelegd op 95. In het volgende hoofdstuk zullen we echter zien dat er om een zeer praktische reden toch een ander adres wordt gekozen.


2 Interface-hardware

De controlesignalen

Met deze fraaie Nederlandse hoofdstuktitel starten we het eerste elektronica-project.


De opzet is erg rechtlijnig. Voor het activeren van de interface hebben we in het vorige hoofdstuk de benodigde signalen achterhaald.

Om een waarde op de DATA-lijnen binnen te halen moeten IOREQ en RD beide hoog zijn. De detectie hiervan kan eenvoudig met een AND-poort, afbeelding 2.1. De uitgang is hoog als beide ingangen hoog zijn.




Afbeelding 2.1 De AND-poort.

Om een waarde op de DATA-lijnen naar buiten te brengen, moeten IOREQ en WR beide hoog zijn. Hiervoor kan eenzelfde schakeling worden gebruikt (afb. 2.2). Een combinatie van deze twee schakelingen (afb. 2.3) is voor de hand liggend.




Afbeelding 2.2 De AND-poort om DATA naar buiten te brengen.



Afbeelding 2.3 Een combinatie van AND-poorten voor het creëren van lees/schrijflijnen.


Er is echter nog een voorwaarde waaraan moet worden voldaan. De adreslijn A6 wordt in een door ons te schrijven toepassingsprogramma hoog gemaakt met een IN- of OUT-opdracht. Alleen als A6 hoog is, mag de interface worden geactiveerd.

Met twee extra poortjes kunnen we dit doel bereiken (afb. 2.4).



Als we echter naar het printaansluitschema van de Spectrum kijken, zien we dat er een streepje boven RD, WR en IOREQ staat. Dat wil zeggen dat deze signalen geïnverteerd zijn. In plaats van AND-poorten zouden we NOR-poorten toe kunnen passen. Om met succes NOR-poorten met drie ingangen te kunnen gebruiken (een uitgang is laag als één van de ingangen hoog is) zou het handig zijn als A6 niet hoog, maar juist laag zou worden. In dat geval doet het schakelingetje volgens afbeelding 2.5 met NOR-poorten de gewenste detectie.

Afbeelding 2.5 De detectieschakeling met NOR-poorten.



We kunnen natuurlijk zeer eenvoudig realiseren dat A6 laag is door een waarde voor het adres te kiezen waarbij A0 t/m A4 hoog zijn, maar A6 laag. Het adres met de laagste waarde is dan 31 (1+2+4+8+16). Een gemakkelijk te onthouden getal. Bij het gebruik van adres 31 doet het er eigenlijk ook niet meer toe of we nu A6, A7 of een andere adreslijn benutten. Bij adreswaarde 31 zijn immers alle lijnen vanaf A4 laag. Dit zou iemand op de gedachte kunnen brengen om de adresdetectie maar helemaal te vergeten en uitsluitend RD, WR en IOREQ als signalering te gebruiken. Een gedachte om snel te vergeten. We beschouwen het schakelgebeuren hier als een statisch geheel. In werkelijkheid zijn ook de schakel-

volgorde en timing van de diverse signalen van belang. De adreslijn speelt in dit opzicht een onmisbare rol.

De opdrachten die binnen een BASIC-programma van de interface kunnen lezen en naar de interface kunnen schrijven, luiden nu respectievelijk:


IN 31
OUT 31, n

Hierin is n een getal tussen 0 en 255 dat we op de interface-uitgang willen hebben.

De DATA-lijnen

Nu we hebben gezien hoe we met enkele poortjes kunnen detecteren wat er van de interface wordt verwacht (er staat nu een software matig te besturen lees- en schrijfsignaal tot onze beschikking) rest nog het aansluiten van de acht DATA-lijnen D0 t/m D7.

Het complete schema van de interface (afb. 2.6) laat zien hoe dit gebeurt.




Afbeelding 2.6 Het schema van de interface.

Om het voor verdere verwerking geschikt te maken wordt het leessignaal eerst geïnverteerd en dan toegevoerd aan een sturingang van het buffer-IC SN74LS244. Dit IC schakelt op commando de acht ingangen door naar de acht uitgangen. De uitgangssignalen zijn rechtstreeks met de Spectrum verbonden. De ingangslijnen van dit IC liggen via weerstanden van 10kΩ aan de voedingspanning. Hierdoor zijn ze in rust hoog. Zoals bij de adressering het geval was, is ook de ingang binair. Als de ingang niet wordt aangestuurd staat er dus het getal 255 op de lijnen.

Lijn	0	1	2	3	4	5	6	7
Binaire waarde	2^0	2^1	2^2	2^3	2^4	2^5	2^6	2^7
Decimale waarde	1+	2+	4+	8+	16+	32+	64+	128 = 255

Voor elke lijn die met massa wordt verbonden, moeten we de bijdrage van 255 aftrekken.

Als voorbeeld nemen we een rij van acht schakelaars (zie afbeelding 2.7) die aan één zijde met massa zijn verbonden, terwijl de andere zijde met een ingangslijn is verbonden. Door steeds weer andere combinaties te sluiten, kan elk getal tussen 0 (alle schakelaars gesloten) en 255 (alle schakelaars open) worden ingesteld.



Afbeelding 2.7 Met acht schakelaars kunnen alle getallen tussen 0 en 255 worden ingesteld.

Het BASIC-programma

```
10 LET p=IN 31
20 PRINT p
```

zal bij de getekende stand van de schakelaars het getal 45 op het scherm zetten. Verderop in dit boek, bij de bespreking van de diverse toepassingen, komt deze codering nog uitgebreid aan de orde.

Zoals het schema in afbeelding 2.6 laat zien, vertoont de uitgang van de interface een grote overeenkomst met de ingang. Het gebruikte IC, een SN74LS373, bevat de nodige elektronica om na het WRITE-commando dat op de enable-ingang wordt gezet, de informatie op de Spectrum DATA-lijnen naar de uitgang door te









































geven en vast te houden tot er een volgend schrijfsignaal verschijnt. Ook de uitgang is binair.

OUT 31, 0 zorgt ervoor dat alle uitgangen 0 zijn.

OUT 31, 255 maakt alle uitgangen hoog.

Als we op de uitgangen denkbeeldige lampen aansluiten, wat stelt dan het volgende programma voor?

```
10 OUT 31, 73
20 OUT 31, 145
30 OUT 31, 36
40 GOTO 10
```

Lampnummer	0	1	2	3	4	5	6	7
73								
145								
36								
73								
145								

enz.

Inderdaad, een looplicht. Een erg snel lopend licht overigens. Verderop in dit boek zullen we een paar disco-effecten bespreken die het bovenstaande programma als uitgangspunt hebben.


In feite hebben we nu een interface tot onze beschikking die op een print kan worden gezet en kan worden aangesloten op de uitbreidingsadapter van de Spectrum. Laten we hier echter nog even mee wachten.

In het volgende hoofdstuk wordt een eenvoudige uitbreiding beschreven waar we straks veel plezier van kunnen hebben.

3 Een analoog uitgangssignaal

Het is natuurlijk aardig een uitgang op onze Spectrum te hebben waar we in binaire vorm getallen op kunnen zetten. Het looplichtprogramma gaf al aan dat daar interessante dingen mee te doen zijn. Veel gebeurtenissen in onze wereld spelen zich echter analoog af. De hoogte van een toon, het toerental van een motor, geluidsterkte, licht, temperatuur en wat al niet meer. Eigenlijk is een analoge uitgangsspanning die vanuit een BASIC-programma te sturen is, een noodzakelijke voorziening.

Gelukkig kan een regelspanning op eenvoudige wijze van het binaire uitgangssignaal worden afgeleid. Voor dit doel passen we een zogeheten R-2R-netwerk toe. Afbeelding 3.1 illustreert een en ander.




Afbeelding 3.1 Het R-2R-netwerk.

De spanning op de uitgang is recht evenredig met de waarde van het decimale getal dat in binaire vorm aanwezig is. De afgeleide regelspanning beweegt zich ongeveer tussen massa en voedingsspanning. Er schuilt echter een klein probleem in de vereiste weerstandswaarden. In de gestandaardiseerde weerstandsreeks komen geen waarden voor die zich als 1 tot 2 verhouden. De oplossing voor dit probleempje is te omzeilen door voor de waarde $2R$ bijvoorbeeld $10k\Omega$ te kiezen. De waarde R ontstaat dan door twee $10k\Omega$ weerstanden parallel te nemen zodat een weerstandswaarde van $5k\Omega$ ontstaat.

Met de toevoeging uit afbeelding 3.2 is het interface-schema compleet. Althans voorlopig. Met de opbouw van de print zullen we rekening houden met uitbrei-

dingen die simpelweg kunnen worden ingeplugd. Uiteraard zal bij die uitbreidingen straks een analoge ingang gaan horen.




Afbeelding 3.2 Het R-2R-netwerk samengesteld met weerstanden van 10k Ω .

4 Het interface-bouwproject

Om een geslaagde uitvoering van het interface-project te waarborgen, is de print bij deze uitgave ingesloten. Hoewel we uitgaan van enkelzijdig printmateriaal en de print liggend achter de Spectrum gebruikt zal worden, kunnen we door een bijzondere uitvoering van de print toch beschikken over de connectoraansluitingen aan de bovenzijde.

Voordat we de stekker van de soldeerbout in het stopcontact steken, zagen we het deel van de aansluitstrip los zoals dat op de print is aangegeven. Zaag ook het vakje uit waar straks de joy-stick-connector wordt gemonteerd. Vervolgens wordt het losgezaagde deel aan de componentenzijde van de hoofdprint vastgezet met soldeerverbindingen, zoals aangegeven in afbeelding 4.1.



Afbeelding 4.1 Het maken van de connector.


Hiertoe wordt een stukje aansluitdraad van een weerstand door de corresponderende gaatjes gestoken en vastgesoldeerd. Mocht dit bij de laatste gaatjes niet lukken omdat de gaten niet voldoende nauwkeurig geboord zijn, haal er dan nog even een boortje door. Nu we toch met het aansluitdeel bezig zijn, solderen we gelijk de connector op zijn plaats. Let er hierbij goed op dat de connector en de print keurig in elkaars verlengde liggen. Steek de connectorpennen niet te ver door. Ze horen uitsluitend verbinding te maken met de soldeeraansluitingen aan de rand. Begin met solderen aan de twee buitenkanten en controleer alvorens verder te gaan of print en connector goed uitgelijnd zijn. De print is nu zover klaar dat we hem op de Spectrum kunnen passen. Kijk goed of de Spectrumaansluitingen voor recorder en voeding nog bereikbaar zijn en of de print vlak achter de computer ligt. Pas eventueel de print wat aan als hij groter blijkt te zijn dan op de transfer aangegeven.

Afwerking

Het is misschien een merkwaardige plaats om iets over de afwerking te zeggen nog voordat er ook maar een onderdeel is aangebracht. Straks zitten echter de onderdelen in de weg en is de hier beschreven activiteit aanzienlijk moeilijker uit te voeren.

Omdat we later vast niet meer weten welke functies bij de verschillende printpennen horen, is het aan te bevelen de componentenzijde van de printplaat te voorzien van opschriften. De lijnnummers van IN- en UITgang, de massa-aansluiting en de voedingsspanningen kunnen met wrijfletters netjes worden aangebracht. Met een grotere letter wordt aangegeven op welk adres de interface aangesproken dient te worden. In dit geval dus 31. Voor het gemak is ook voor dit karwei een wrijftransfer opgenomen.

Het moeilijkste deel is nu achter de rug en de print kan worden bestukt. Hoewel de meeste bouwbeschrijvingen adviseren te beginnen met het plaatsen van de draadbruggen, is er geen enkel bezwaar tegen om eerst de IC-voeten en de weerstanden te monteren. Het lijkt dan al gauw wat en bovendien kunnen de afgeknipte aansluitdraadjes als draadbrugjes worden benut! Voor de langere draadbruggen is het verstandig geïsoleerd draad te nemen. De kans op sluitingen wordt daardoor aanzienlijk kleiner. Let erop dat de parallel geschakelde weerstanden van het R-2R-netwerk rechtop worden gemonteerd en wel om en om



Afbeelding 4.2 De compleet gemonteerde interfaceprint. Let op de om en om gemonteerde weerstanden.

zoals de foto aangeeft. Ook dit weer in verband met onverhoopt contact tussen de diverse onderdelen.

Alle weerstanden zijn $10k\Omega$. Voor de printpennen, die als laatste worden aangebracht, nemen we een uitvoering waar stekkertjes bij worden geleverd. Zorg ervoor dat ze keurig rechtop staan.


Voor we de print op de Spectrum aansluiten voorzien we een draadje aan weerszijden van een stekkertje dat op de printpennen past.

De interface is nu klaar voor gebruik. Omdat hij geen doorverbinding heeft naar een volgende connector, sluiten we hem aan achter eventuele andere randapparatuur (uitbreidingen, printer e.d). Schakel altijd de voedingsspanning uit bij het inpluggen of verwijderen van apparatuur!

Achterin het boek zijn de lay-outs opgenomen van de hulpprinten die worden besproken. Men kan deze lay-outs gebruiken voor het zelf vervaardigen van de printen.

5 Uitgangsmonitor


In een later stadium zal de binaire interface-uitgang worden gebruikt om er lampen en andere elektrische verbruikers mee te schakelen. Om het ontwikkelen van de hiertoe benodigde programmatuur te vereenvoudigen wordt in dit hoofdstuk een handig programmeerhulpje beschreven. Elektronisch stelt het niet veel voor. Acht LEDjes met de bijbehorende stroombegrenzingsweerstand. Eén en ander op een handzaam printje dat zo op de hoofdprint kan worden gestoken. Voor dit printje is de lay-out achterin opgenomen zodat het probleemloos kan worden nagebouwd. Let goed op de stand van de LED's (afb. 5.1).



Afbeelding 5.1 Penaansluiting interface.

De print kan zonder bezwaar worden aangebracht terwijl de hoofdprint in bedrijf is. Hij past maar op één manier op de printpennen. Raak bij deze manoeuvre echter geen andere pennen aan. Neem desnoods het zekere voor het onzekere en schakel de voeding even uit. De opgestoken print kan naar wens iets achterover worden gebogen om het zicht op de LED's te verbeteren (afb. 5.2).

Met deze uitgangsmonitor is het testen van de hoofdprint snel gedaan zoals uit het volgende hoofdstuk zal blijken.



Afbeelding 5.2 De interface met de opgestoken uitgangsmonteur.

6 Test

Als eerste testen we de ingang.

Hiertoe typen we het volgende programma in:

```
1Ø LET p=IN 31
2Ø PRINT AT Ø,Ø;p
3Ø GOTO 1Ø
```

Bij het uitvoeren van dit programma wordt het decimale getal dat op de ingang wordt aangeboden linksboven op het scherm weergegeven.

Omdat alle lijnen via een weerstand aan de voeding liggen zijn ze hoog. Hieruit volgt dan dat het getal 255 zichtbaar moet zijn.

Verbind nu het van stekkertjes voorziene draadje met de massapen en raak met de andere kant één voor één de ingangspennen aan. De corresponderende getallen worden dan zichtbaar. Probeer ook eens meerdere pennen tegelijk aan te raken, ook dan zien we het corresponderende getal verschijnen.

Tot zover de ingang.

Ook de uitgang kunnen we met een kort programma testen.

```
1Ø PRINT AT Ø,Ø;"geef getal tussen Ø en 255"
2Ø INPUT p
3Ø OUT 31,p
4Ø PRINT AT 2,Ø;p;"[ ][ ]"
5Ø GOTO 2Ø
```

Met een universeelmeter kunnen we nu op de uitgangssprintpennen controleren of het ingegeven getal in binaire vorm is verschenen.

De twee spaties in regel 40 wissen het restant van een vorige uitlezing als die meer cijfers had dan het nieuw getal.

In plaats van de universeelmeter is het natuurlijk veel handiger om de uitgangsmonitor op de hoofdprint te prikken. Het voorgaande programma laat dan naar wens één of meer LED's oplichten.

Een programma waarmee we de LED's één voor één als een soort lichtslang op laten lichten, zou er als volgt uit kunnen zien:

```

10 REM lichtslang
20 FOR i=0 TO 7
30 OUT 31,2^i
40 NEXT i
50 GOTO 10

```

Het effect is weliswaar aardig, maar de snelheid waarmee de slang zich beweegt, is niet erg indrukwekkend. Het door Sinclair toegepaste BASIC is nu eenmaal niet erg snel en het sommetje in regel 30 vraagt een zee van tijd. In plaats van de computer twee te laten verheffen tot de macht 0, 1, 2 enz. kunnen we dat beter zelf doen:

```

10 REM snelle slang
20 OUT 31,1: OUT 31,2: OUT 31,4: OUT 31,8
30 OUT 31,16: OUT 31,32: OUT 31,64: OUT 31,128
40 GOTO 10

```

Dit resultaat ziet er wat snelheid betreft meer bevredigend uit. Het kan dus echt wel snel. De snelheid is te regelen door tussen elke OUT-opdracht een PAUSE-commando op te nemen.

Ook eenvoudig is de test van het R-2R netwerk. Sluit een universeelmeter (bereik 5 volt of meer) aan op de analoge uitgang en toets het volgende programma in.

```

10 FOR i=0 TO 255
20 OUT 31,i
30 PAUSE 10
40 NEXT i
50 GOTO 10

```

Dit programma laat de waarde van i oplopen van 0 tot 255. De uitgangsspanning van de digitaal/analoog omzetter zal dan ook oplopen van ongeveer 0 tot de maximale waarde.

HET BEREIK LOOPT VAN 0,1 TOT 3,5 VOLT

Regel 50 zorgt ervoor dat de cyclus zich herhaalt. Door het getal in het PAUSE-commando te veranderen, kan de frequentie worden gewijzigd. De elektronicus zal al snel inzien dat we zojuist een softwarematig bestuurd zaagtandgenerator hebben gecreëerd. In de hoofdstukken 17 en 18 zullen we de programmatuur uitbreiden tot een generator waarmee we dubbele sinussen en in fase verschoven sinussen kunnen genereren. Omdat er volledige softwarecontrole is, kunnen we uiteraard ook allerlei berekende functies in de vorm van een analoge spanning weergeven en zonodig cyclisch herhalen.


7 Joy-stick

Nu we over een binaire ingang beschikken, vormt de zelfbouw van een joy-stick vanzelfsprekend een uitdaging. Er is veel commerciële software verkrijgbaar die speciaal voor een joy-stick is geschreven. Bovendien is de joy-stick een bijzonder veelzijdig instrument omdat deze een mens/machine-interface is die ons natuurlijk gevoel voor beweging en richting koppelt aan de reken- en tekensnelheid van een computer.

De Spectrum-bezitter die begonnen is met de ZX81 kent wellicht het boek 'Zelfbouw-joy-stick voor de ZX81' van dezelfde uitgever. Hierin wordt uitgebreid ingegaan op de bouw en het gebruik van een joy-stick. De vele programma's in dit werk laten zien dat het gebruik van een joy-stick ver boven de toepassing in spelletjes uit kan gaan. Met name het gebruik als 'muis' is erg intrigerend. In dit hoofdstuk zal in zeer kort bestek de bouw en het gebruik van een joy-stick worden beschreven. Voor een meer uitgebreide behandeling wordt verwezen naar het desbetreffende ZX81-boekje.

Een joy-stick bestaat in principe uit vijf schakelaars waarvan er vier in ruitvorm zijn opgesteld. De vijfde wordt als vuurknop gebruikt.

Een eenvoudige maar uiterst veelzijdige joy-stick is uitgewerkt in afbeelding 7.1.




Afbeelding 7.1 Joy-stick bestaande uit vier miniatuur-drukschakelaars.

Vier miniatuur-drukschakelaartjes (maakcontact) worden in ruitvorm in een passende behuizing gemonteerd. Eén aansluiting wordt met de massa van de interfaceprint verbonden, terwijl de andere aansluitingen met de ingangslijnen 0 t/m 3 worden verbonden. Dit kan het beste gebeuren in de volgorde zoals die bij veel in de handel zijnde joy-sticks wordt gehanteerd.

De vuurknop wordt verbonden met lijn 4.

Voor het bedienen van de schakelaars brengen we een centrale bout aan waarop een ring tussen twee moeren wordt bevestigd. De hoogte van de ring wordt zodanig afgesteld dat er juist geen speling meer bestaat tussen ring en schakelaars. Op het uitstekende boutende kan naar eigen voorkeur een bedieningshandel worden geschroefd of gelijmd.

Voor de vuurknop kan het beste een wat robuustere drukknop worden genomen.




Afbeelding 7.2 De joy-stick wordt op de koperzijde aangesloten.

Omdat de interface voor vele doeleinden kan worden gebruikt en het nogal onhandig is als er altijd een joy-stick aan vast gesoldeerd zit, is het raadzaam de verbinding tot stand te brengen via een connector. Wellicht bent u al in het bezit van een joy-stick, of ziet u op tegen het zelf bouwen ervan. Het kan ook zijn dat u per se dat ene speciale 'Arcade action'-type wilt toepassen. Kortom, de aansluitplug vraagt om enige standaardisatie voor zover dat in de computerwereld mogelijk is. Een veel gebruikte aansluiting is de negenpolige D-connector. Het mannelijke chassisdeel dat we nodig hebben, is vrijwel overal voor enkele gulden te koop. Door een chassisdeel te nemen bent u vrij de connector te monteren op de meest geschikte plaats. Hij kan ook direct op de print worden gemonteerd en wel op het speciaal daarvoor gereserveerde gedeelte. Om de print niet nodeloos ingewikkeld te maken, worden de chassisdeelpennen met korte draadjes met de overeenkomstige ingangspennen verbonden. Dit kan het beste aan de koperzijde gebeuren zoals afbeelding 7.2 aangeeft. Op deze wijze houden we de pennen zelf vrij voor andere toepassingen.

Het is overigens geen enkel bezwaar om de joy-stick gewoon aangesloten te laten terwijl de ingangspennen voor een ander doel worden gebruikt. In rust staan alle joy-stick-schakelaars immers open en hebben dan ook geen enkele invloed. Om enig idee te geven van de mogelijkheden van een joy-stick zullen we enkele

tekenprogramma's uitwerken. Terwille van het speelse karakter van de joy-stick komen er ook enkele bouwstenen voor spelletjes aan de orde.



Afbeelding 7.3 Koperzijde hoofdprint.

8 Ezeltje prik

Alvorens het programma te bespreken, is het raadzaam het hieronder afgedrukte programma te bekijken. Het beeldt onze nationale driekleur af door drie woorden (strings) af te drukken. De lengte van deze woorden (elk woord bestrijkt ongeveer eenderde van het scherm) is vastgelegd in een DIM-opdracht. Het woord bestaat uit spaties met een bepaalde PAPER-kleur. Na een korte pauze wordt er een nieuw woord op het scherm geschreven, ditmaal met een lengte van 704 gekleurde spaties (dus precies een scherm vol). Na een nieuwe pauze herhaalt het proces zich. Deze techniek om zeer snel een deel van het beeld in te kleuren, zult u in dit boek regelmatig aantreffen.

```
1  REM driekleur
10 BORDER 5

20 DIM b$(224)
30 PRINT AT 0,0; PAPER 2;b$
40 PRINT AT 7,0; PAPER 7;b$
50 PRINT AT 14,0; PAPER 1;b$
60 PAUSE 100

80 DIM c$(704)
90 PRINT AT 0,0; PAPER 5;c$
100 PAUSE 50
110 GOTO 20
```

In het 'Ezeltje prik'-programma wordt de joy-stick als potlood gebruikt. Het is de bedoeling om de ezel die op het scherm verschijnt van een passende staart te voorzien. Uiteraard dient dat geblinddoekt te gebeuren. Omdat een echte blinddoek niet erg goed in onze informatiemaatschappij past, gebruiken we een schermtruc. De staart wordt getekend in de kleur van de achtergrond. Hij wordt pas zichtbaar als de vuurknop wordt ingedrukt. Behalve de allereerste keer. Die mag worden benut om even te oefenen.

```
1  REM ezeltje prik
2  CLS: INK 6: PAPER 4: BORDER 4
```

```

10 CIRCLE 100,70,40
20 CIRCLE 60,110,25
30 CIRCLE 50,120,4
40 INK 2
50 PLOT 37,115
60 DRAW -25,5
70 DRAW 24,-10
80 DRAW -25,-5
90 DRAW 23,0
100 PLOT 105,30
110 DRAW 0,-25
120 DRAW -25,0
130 DRAW 0,-5
140 DRAW 40,0
150 DRAW 0,5
160 DRAW -18,0
170 DRAW 0,25
180 PLOT INK 6; 80,80
190 DRAW INK 6; 20,-30,3
200 DRAW INK 6; 30,25,.5
210 CIRCLE INK 5; 49,118,1
220 CIRCLE INK 5; 49,118,2
230 PLOT INK 4; 135,95
240 PRINT AT 0,0;"Tekenen een staart"
250 PRINT "en druk daarna op de knop"
260 BEEP .3,20: BEEP .3,30: BEEP .3,20

270 LET p=IN 31
280 LET x=0: LET y=0
290 IF p=254 OR p=250 OR p=246 THEN LET y=y+1
300 IF p=253 OR p=249 OR p=245 THEN LET y=y-1
310 IF p=251 OR p=250 OR p=249 THEN LET x=x-1
320 IF p=247 OR p=246 OR p=245 THEN LET x=x+1
330 IF p=239 THEN GOTO 900
340 DRAW INK 4;x,y
350 GOTO 270

900 DIM b$(704)
910 PRINT AT 0,0;"Nog een keer: druk knop"
920 PRINT "Stop : beweeg stick"
930 PRINT AT 0,0; OVER 1; b$
940 LET p=IN 31
950 IF p=239 THEN GOTO 2

```

```

96Ø IF p<>255 THEN STOP
97Ø GOTO 94Ø

```

De tweede regel legt de schermkleuren vast. In het daarop volgende grote blok wordt de ezel getekend. De laatste regel van dit blok (230) zet de potloodpunt op de plaats waar de staart moet beginnen. Het blok vanaf regel 270 kijkt of de joy-stick is bediend en verandert zonodig de x- en y-coördinaat van de potloodpunt. Regel 330 kijkt of de vuurknop is ingedrukt. Als dat het geval is, komt het laatste blok in actie. De schermkleur (PAPER) wordt veranderd waardoor de zojuist getekende staart zichtbaar wordt.

Het belangrijke deel van dit programma staat in de regels 290 t/m 320.

```

29Ø IF p=254 OR p=25Ø OR p=246 THEN LET y=y+1
3ØØ IF p=253 OR p=249 OR p=245 THEN LET y=y-1
31Ø IF p=251 OR p=25Ø OR p=249 THEN LET x=x-1
32Ø IF p=247 OR p=246 OR p=245 THEN LET x=x+1

```

In regel 290 wordt de y-coördinaat met 1 verhoogd als de stick naar voren is geduwd. Echter ook als de stick schuin naar voren wordt geduwd, gaat de waarde van y omhoog. In dat geval zijn er twee schakelaars ingedrukt. In de vorige hoofdstukken is de codering al uitgebreid aan de orde geweest.

Het Sinclair BASIC laat een iets compactere schrijfwijze toe voor meervoudige statements. We kunnen regel 290 bijvoorbeeld ook als volgt schrijven:

```

29Ø IF (p=254) OR (p=25Ø) OR (p=246) THEN LET y=y+1

```

De uitdrukking (p=254) wordt door de computer wel heel erg logisch bekeken. Als p=254 waar is, dan is (p=254) gelijk aan 1.

Is p=254 niet waar, dan geldt (p=254)=0.

Met deze kennis kunnen we het programmablokje als volgt schrijven:

```

29Ø LET y=y+(p=254)+(p=25Ø)+(p=246)-(p=253)-(p=249)-
      (p=245)
31Ø LET x=x+(p=247)+(p=246)+(p=245)-(p=251)-(p=25Ø)-
      (p=249)

```

De regels 290 en 300 zijn nu gecombineerd. Y wordt verhoogd met 1 als de stick naar voren of schuin naar voren wordt geduwd en verlaagd als de stick naar achteren of schuin naar achteren wordt getrokken. Voor de horizontale beweging geeft regel 310 nu alle mogelijke combinaties weer.

Deze compacte schrijfwijze zal ook in nog volgende programma's worden gebruikt.

Vindt u de vorm van de ezel ietwat merkwaardig? Ik daag u uit een betere te tekenen, uiteraard met de joy-stick! Hoe dat moet, leert het volgende programma.

9 Tekenen met de joy-stick

Uit 'ezeltje prik' is duidelijk geworden hoe met de joy-stick tekeningen met een hoog oplossend vermogen kunnen worden gemaakt. Precies dezelfde techniek is in het hier gegeven programma gebruikt om een tekenplank te maken. Uiteraard zijn er enkele extra's aan toegevoegd. Zo kunnen we de kleuren van de rand (border), het tekenvlak (paper) en het potlood (ink) kiezen. Bovendien is er de keuze uit vier snelheden.

```
10 REM tekenen
20 LET x=0: LET y=0
30 CLS: BORDER 5: PAPER 7: INK 0

40 PRINT AT 2,15; FLASH 1;"ACHTERGROND"
50 GOSUB 8000
60 PAUSE 10
70 LET papier=k
80 CLS

90 PRINT AT 2,20; FLASH 1;"LIJN"
100 GOSUB 8000
110 PAUSE 10
120 LET inkt=k
130 CLS

140 PRINT AT 2,20; FLASH 1;"RAND"
150 GOSUB 8000
160 PAUSE 10
170 LET rand=k
180 CLS

190 PRINT AT 2,2; FLASH 1;"KIES SNELHEID"
200 PRINT AT 5,5;"langzaam"
210 PRINT AT 7,5;"vlugger"
220 PRINT AT 9,5;"snel"
230 PRINT AT 11,5;"zeer snel"
```

```

240 LET l=5
250 LET p=IN 31
260 LET l=1+2*((p=253)-(p=254))
270 IF l<5 THEN LET l=5
280 IF l>11 THEN LET l=11
290 LET s=1-4
300 PRINT AT 1,3; INK 2; "■"
310 PAUSE 3
320 PRINT AT 1,3; OVER 1; "■"
330 IF p=239 THEN GOTO 350
340 GOTO 250

350 PAPER papier: INK inkt: BORDER rand
360 PAUSE 10 6
370 CLS
380 LET p=IN 31
390 LET x=x+s*((p=247)+(p=246)+(p=245)-(p=251)-
    (p=250)-(p=249))
400 LET y=y+s*((p=254)+(p=250)+(p=246)-(p=253)-
    (p=249)-(p=245))
410 IF p=239 THEN GOTO 900
420 IF x<0 THEN LET x=0
430 IF x>255 THEN LET x=255
440 IF y<0 THEN LET y=0
450 IF y>175 THEN LET y=175
460 PLOT x,y
470 GOTO 380

900 PAPER 7: INK 0: BORDER 7
910 STOP

8000 PRINT AT 2,4;"KIES KLEUR"
8010 PRINT AT 4,7;"zwart"
8020 PRINT AT 5,7;"blauw"
8030 PRINT AT 6,7;"rood"
8040 PRINT AT 7,7;"paars"
8050 PRINT AT 8,7;"groen"
8060 PRINT AT 9,7;"licht blauw"
8070 PRINT AT 10,7;"geel"
8080 PRINT AT 11,7;"wit"
8090 LET l=4
8100 LET p=IN 31
8110 LET l=1+(p=253)-(p=254)

```

```

8120 IF 1<4 THEN LET 1=4
8130 IF 1>11 THEN LET 1=11
8140 PRINT AT 1,4; INK 2; "■"
8150 PAUSE 3
8160 PRINT AT 1,4; OVER 1; "■"
8170 LET k=1-4
8180 IF p=239 THEN RETURN
8190 GOTO 8100

```

Aan het begin van het programma staan drie blokjes waarmee de te gebruiken kleuren kunnen worden vastgelegd. Ze maken alle drie gebruik van de subroutine die in de twee blokken vanaf regel 8000 is opgenomen. In deze subroutine krijgt de variabele k een waarde die correspondeert met de gekozen kleur. De waarde van k wordt afgeleid van de variabele die het regelnummer aangeeft waarop de cursor stond op het moment dat in regel 8180 de vuurknop was ingedrukt. De plaats van de cursor kan worden veranderd door de joy-stick voor- of achterwaarts te bewegen. Regel 8110 laat de constructie zien zoals die in het vorige hoofdstuk is omschreven.

Om te voorkomen dat de cursor buiten de aangewezen plaatsen terechtkomt, is in de regels 8120 en 8130 een beveiliging ingebouwd.

Dezelfde techniek is toegepast in het blok dat op regel 190 begint. In dit blok wordt de tekensnelheid gekozen. Omdat deze kiesprocedure maar één keer voorkomt, is er geen subroutine voor gemaakt.

Het eigenlijke tekenen gebeurt in het blok met regelnummers vanaf 380. De nieuwe x- en y-positie van de pixel worden direct uitgePLOT.

Een tekenprogramma als dit vraagt om verdere uitbreiding. Het uitvlakken van lijnen, inkleuren van vlakken en tussentijds veranderen van kleuren zijn mogelijkheden waarmee de tekenplank kan worden uitgebreid tot een echt grafisch gereedschap.

Als de tekening bevalt, kan hij op tape worden bewaard door hem te 'saven' met het commando:

```
SAVE"tekening"SCREEN$
```

De tekening kan dan vanuit elk ander programma worden opgeroepen met:

```
LOAD"tekening"SCREEN$
```

Hoofdstuk 20 van het Engelstalige handboek waarschuwt ervoor dat bij het gebruik van SAVE"naam"SCREEN\$ de verificatieprocedure VERIFY"naam" niet werkt. Voor unieke tekeningen waar veel werk in zit, is het dan ook beter de opdracht:

SAVE"tekening"CODE 16384,6912

te gebruiken. De getallen geven respectievelijk het eerste geheugenadres en de lengte aan. Met de bovenstaande getallen past er een volledig beeldscherm in de gereserveerde ruimte zonder dat er belangrijke geheugenplaatsen ongewild van foutieve informatie worden voorzien. Het laden van een tekening gebeurt in dit geval met

LOAD"tekening"CODE

Hier mogen de getallen worden weggelaten.

10 Pixel-restanten

Als we de Spectrum opdracht geven een lijn op het scherm te trekken met de DRAW-opdracht, worden er pixels ingekleurd. Willen we de lijn weer wissen dan kan dat met de opdracht DRAW OVER 1. Hierbij is het belangrijk dat er van hetzelfde beginpunt wordt uitgegaan.

Het volgende programma trekt een lijn vanaf de pixel-coördinaten 10,20 naar 50,70. De lijn blijft een ogenblik staan en wordt dan netjes gewist.

```
10 PLOT 10,10
20 DRAW 200,130
30 PAUSE 50
40 PLOT 10,10
50 DRAW OVER 1;200,130
```

Ogenschoijnlijk doet het volgende programma hetzelfde.

```
10 PLOT 10,10
20 DRAW 200,130
30 PAUSE 50
50 DRAW OVER 1;-200,-130
```

De lijn wordt getrokken, maar nu in tegengestelde richting gewist. Op het scherm blijven er dan hier en daar pixels staan. Er zit een bepaalde logica achter dit verschijnsel. Voor elke plaats op het scherm blijven er weer andere pixels staan. In het hierna afgedrukte programma wordt dankbaar gebruik gemaakt van dit verschijnsel. Door met de joy-stick een pixel over het scherm te dirigeren, kunnen er interferentiepatronen worden opgeroepen.

```
1 REM pixel restanten
2 PAPER 4: INK 2
10 LET x=0: LET y=0
20 PLOT 0,0
30 DRAW x,y
40 DRAW OVER 1;-x,-y
50 LET p=IN 31
```

```

60 LET x=x+(p=247)+(p=246)+(p=245)-(p=251)-(p=250)-
  p=249)
70 LET y=y+(p=254)+(p=250)+(p=246)-(p=253)-(p=249)-
  p=245)
80 IF p=239 THEN GOTO 900
90 IF x<0 OR x>255 OR y<0 OR y>175 THEN GOTO 50
100 PLOT x,y
110 GOTO 20

900 PAPER 7: INK 0
910 STOP

```

Regel 900 zorgt ervoor dat na het beëindigen van het programma het scherm weer een witte achtergrond met zwarte tekst heeft.

Verander voor de aardigheid de regels 40 en 90 eens in

```

40 DRAW OVER1;x,y
90 IF x<0 OR x>125 OR y<0 OR y>85 THEN GOTO 50

```

Er ontstaan dan zeer fraaie interferentiepatronen.

11 Spokenjacht

Dat we met de joy-stick niet alleen puntjes op het scherm kunnen laten bewegen, maar ook samengestelde figuren (of teksten) kunnen we het beste waarnemen in een spel. U rijdt in een speciaal voertuig op de snelweg. Voor u doemen echter met een angstaanjagende snelheid spoken op die kennelijk de jacht op "Packman" hebben opgegeven. U kunt ze ontwijken. U kunt ze ook vangen met een voor dit doel ontwikkelde antenne. Dat levert bovendien punten op.

```
1 REM spokenjacht
10 LET n=0: LET x=10: LET m=0
20 RANDOMIZE
30 GOSUB 8000
40 DIM a$(224)
50 PRINT AT 0,0; OVER 1; PAPER 5;a$
60 DIM b$(480)
70 PRINT AT 7,0; OVER 1; PAPER 6;b$
80 PLOT 0,120: DRAW 255,0
90 LET r= RND
100 FOR i=1 TO 21
110 INK 0
120 PLOT 132-i*i/4,119-i*i/4
130 PLOT 132+i*i/4,119-i*i/4
140 NEXT i

150 PRINT AT 18,x+4; PAPER 6;"Y"
160 PRINT AT 19,x+1;" █ █ █ █ █ "
170 PRINT AT 20,x;" █ █ █ █ █ █ █ "
180 PRINT AT 21,x+1;" █ █ █ █ █ "
190 FOR i=1 TO 21
200 BEEP .01,20
210 LET p=IN 31
220 IF p=239 THEN STOP
230 LET a=x
240 LET x=x+(p=247)-(p=251)
250 IF x<0 OR x>25 THEN LET x=a
```

```

260 IF a=x THEN GOTO 350
270 PRINT AT 18,a+4; OVER 1; PAPER 6;"Y"
280 PRINT AT 19,a+1; OVER 1; PAPER 6;"■ ■ ■ ■ ■"
290 PRINT AT 20,a; OVER 1; PAPER 6;"■ ■ ■ ■ ■"
300 PRINT AT 21,a+1; OVER 1; PAPER 6;"■ ■ ■ ■ ■"
310 PRINT AT 18,x+4; PAPER 6;"Y"
320 PRINT AT 19,x+1;"□ □ □ □ □"
330 PRINT AT 20,x;"■ ■ ■ ■ ■"
340 PRINT AT 21,x+1;"■ ■ ■ ■ ■"

350 IF i<4 THEN PRINT AT 7+i,16; PAPER 6;"A" *)
360 IF i<4 THEN PRINT AT 7+i,16; OVER 1; PAPER 6;"A" *)
370 IF i>4 AND i<8 THEN PRINT AT
    8+i/2,17+r*i*SGN(r-.5)/2; PAPER 6;"B" *)
380 IF i>4 AND i<8 THEN PRINT AT 8+i/2,17+r*i*SGN
    (r-.5)/2; OVER 1; PAPER 6;"B" *)
390 IF i<8 AND i>14 THEN PRINT AT 8+i/2,17+r*i*SGN
    (r-.5)/2; PAPER 6;"C" *)
400 IF i<8 AND i>14 THEN PRINT AT 8+i/2,17+r*i*SGN
    (r-.5)/2; OVER 1; PAPER 6;"C" *)
410 LET w$=SCREEN$ (8+i/2,17+r*i*SGN (r-.5)/2)
420 IF w$="Y" THEN GOSUB 9000
430 IF i>14 THEN PRINT AT 8+i/2,17+r*i*SGN (r-.5)/2;
    PAPER 6;"D" *)
440 IF i>14 THEN PRINT AT 8+i/2,17+r*i*SGN (r-.5)/2;
    OVER 1; PAPER 6;"D" *)
450 LET w$=SCREEN$ (8+i/2,17+r*i*SGN (r-.5)/2)
460 IF w$="Y" THEN GOSUB 9000
470 NEXT i

480 LET m=m+1
490 PRINT AT 2,28; PAPER 5;m
500 GOTO 90

8000 RESTORE
8010 FOR a=USR "a" TO USR "a"+7
8020 READ bite
8030 POKE a,bite
8040 NEXT a
8050 FOR b=USR "b" TO USR "b"+7
8060 READ bite
8070 POKE b,bite

```

```


8080 NEXT b
8090 FOR c=USR "c" TO USR "c"+7
8100 READ bite
8110 POKE c,bite
8120 NEXT c
8130 FOR d=USR "d" TO USR "d"+7
8140 READ bite
8150 POKE d,bite
8160 NEXT d
8170 DATA 0,0,0,24,24,0,0,0,0,0,60,60,60,36,0,0,0,0,
          126,90,126,90,90,0,56,124,214,214,254,254,
          170,170
8180 RETURN
8180 RETURN

9000 LET n=n+1
9010 PRINT AT 2,2; PAPER 5;n
9020 BEEP .5,10: BEEP .5,30: BEEP .5,25
9030 RETURN

```

Het eerste blok tekent de horizon en een weg. Het tweede blok laat de afbeelding van een auto verschijnen op de plaats die met de joy-stick kan worden aangegeven. Blok drie roept een grafisch karakter op dat in de subroutine vanaf regel 8000 wordt gedefinieerd. Let erop dat in de regels die met *) zijn gemerkt een grafisch karakter voorkomt. Dat betekent dat met CAPS SHIFT 9 eerst de grafische cursor [d] moet worden opgeroepen. Pas dan wordt de A, B, C of D ingetypt. Door de grafische karakters steeds groter te maken, wordt de indruk gewekt dat de spoken zich naar ons toe bewegen. De kleine subroutine vanaf regel 9000 houdt de puntentelling bij.

Wat betreft de joy-stick-techniek is alleen de beveiliging afwijkend. Regel 250 houdt rekening met de breedte van de auto door de cursor in zijn bewegingsvrijheid te beperken (25 in plaats van 32 posities op één regel).



Afbeelding 11.1
De spokenjacht.

12 Het meten van de tijd

Door gebruik te maken van de binaire ingang is het meten van het tijdsverloop tussen twee gebeurtenissen eenvoudig. We hebben een schakelaar met een maakcontact nodig die we aansluiten op een willekeurig ingangskanaal. Het begin van het proces wordt gegeven door de schakelaar kort in te drukken. Een volgende druk op de knop geeft het einde van het proces aan.

```
1 REM tijdmeting
10 LET p=IN 31
20 IF p=255 THEN GOTO 10
30 IF p<>255 THEN start klok
40 PRINT tijd
```

Invoeren van dit programma is vergeefse moeite, er zal maar weinig gebeuren. Dit voorbeeldje geeft echter wél aan hoe een tijdmeting wordt uitgevoerd. Regel 10 kijkt of één van de interface-ingangen is gesloten. Als er een contact is gesloten, wordt de verstreken tijd op de een of andere manier vastgelegd.

Om op eenvoudige wijze vertrouwd te raken met de programmatuur nemen we ter demonstratie de joy-stick als schakelaar. We maken gelijk gebruik van de mogelijkheid om verschillende schakelaars apart uit te lezen.

```
1 REM tijdmeting
10 PRINT AT 21,0;"STICK hoog laag links rechts"

20 LET p=IN 31
30 IF p<> 254 THEN GOTO 20
40 POKE 23672,0

50 LET p=IN 31
60 IF p<>253 THEN GOTO 50
70 LET t1=PEEK 23672
80 PRINT AT 1,0;"Interval 1: ";t1/50;" sec"

90 LET p=IN 31
```



```

100 IF p<>251 THEN GOTO 90
110 LET t2=PEEK 23672
120 PRINT AT 2,0;"Interval 2: ";t2/50;" sec"

130 LET p=IN 31
140 IF p<>247 THEN GOTO 130
150 LET t3=PEEK 23672
160 PRINT AT 3,0;"Interval 3: ";t3/50;" sec"

```

Dit programma maakt gebruik van een systeemvariabele die vijftig maal per seconde wordt verhoogd. De naam FRAMES van deze variabele geeft aan dat hij de beeldfrequentie van het TV-signaal bijhoudt. FRAMES is te vinden op adres 23672. Aangezien de Spectrum een acht-bits machine is, is het maximale getal dat in geheugenadres 23672 past 255. Omdat FRAMES 50 maal per seconde wordt verhoogd, kunnen we tellen tot $255/50=5,1$ seconden. Daarna springt de adresinhoud weer naar 0 om in 5,1 seconde opnieuw naar de waarde 255 te gaan.

Om op nul te kunnen beginnen 'poken' we een nul naar het desbetreffende adres, zodra de stick naar voren wordt bewogen. Bij het naar achteren trekken van de stick wordt het FRAMES-adres uitgelezen en gedeeld door 50 zodat de tijd in seconden op het scherm kan worden gezet. Hetzelfde gebeurt als de stick naar links en vervolgens naar rechts wordt bewogen. Het is intussen duidelijk dat we op deze wijze nooit verder komen dan ruim vijf seconden. We kunnen natuurlijk door het programma laten bijhouden wanneer FRAMES nul wordt. Als dit bijvoorbeeld drie keer is gebeurd, weten we dat er driemaal 5,1 seconden verstreken zijn. In dat geval moeten we echter constant FRAMES uitlezen. Het volgende programma zal illustreren dat Sinclair hier goed over heeft nagedacht.

13 Stopwatch

In hoofdstuk 12 hebben we gezien dat op adres 23672 de systeemvariabele FRAMES is te vinden. Dat is echter niet de volledige waarheid. In werkelijkheid is het volgende adres, 23673 dus, een soort overloop voor het eerste FRAMES-geheugen. Zodra het eerste adres vol is (255) springt de inhoud naar 0, maar wordt er een 1 gezet in het geheugen op het volgende adres. Een derde geheugenplaats (23674) dient weer als overloop voor het tweede geheugen. Een rekensommetje leert nu dat het maximale getal dat in de drie geheugens kan worden opgeslagen $255 + 256*255 + 256*256*255 = 16777215$ bedraagt, zodat er tijden tot $16777215/50=335544,3$ seconden mogelijk zijn. Dat is bijna vier dagen.

Er bestaat echter een gevaar. Het uitlezen van de afzonderlijke geheugenplaatsen vergt enige tijd. Bij het uitlezen van het laatste adres is de inhoud van het eerste adres alweer veranderd. Het handboek geeft in hoofdstuk 18 een oplossing. Bij deze oplossing wordt gebruik gemaakt van door uzelf vast te leggen functies (User defined functions). In het volgende programma is van deze oplossing gebruik gemaakt. Als stopwatch wordt een schakelaartje met maakcontact toegepast. Het geeft niet op welke ingangslijn u hem aansluit. Als u de joy-stick bij de hand heeft nemen we gewoon de vuurknop. Kort indrukken start de klok. Na elke volgende korte druk verschijnt de verstreken tijd in een tabel op het scherm.

```
1 REM stopwatch
10 GOTO 100

20 REM subroutine tijdmeting
30 DEF FN s()=(PEEK 23672+256*PEEK 23673+65536*PEEK
  23674)/50
40 DEF FN g(a,b)=(a+b+ABS (a-b))/2
50 DEF FN t()=FN g (FN s(),FN s())
60 LET tijd=FN t()
70 RETURN

100 REM hoofdprogramma
110 LET n=0
120 PRINT "TIJDMETING"
```

```

130 FOR i=1 TO 32: PRINT INK 5;"□";: NEXT i
140 PRINT "DEELNEMER TIJD"
150 LET p=IN 31
160 IF p=255 THEN GOTO 150
170 POKE 23672,0: POKE 23673,0: POKE 23674,0
180 PAUSE 5
190 PRINT AT 0,20; INK 2; FLASH 1;" klok loopt"
200 PRINT AT 2,0
210 LET p=IN 31
220 IF p=255 THEN GOTO 210
230 LET n=n+1
240 GOSUB 20
250 PRINT n, INT (tijd*10)/10
260 PAUSE 5
270 GOTO 210

```

Regel 10 zorgt voor een sprong over de subroutine heen waarin de daadwerkelijke tijdmeting plaatsvindt. Let erop dat DEF FN in deze subroutine een enkele opdracht is. In regel 170 wordt de klok op 0 gezet. De opmerkzame lezer zal zich afvragen waar de PAUSE-opdrachten voor dienen. De functie hiervan moge blijken uit het volgende:

Bij het indrukken van de schakelaar wordt het programma verder afgewerkt. Het kan dan gebeuren dat de schakelaar nog ingedrukt is terwijl de volgende tijdmeting nog niet aan de orde is. De PAUSE-opdracht geeft even tijd om de knop na een korte druk weer los te laten. De intervallen kunnen dan ook niet korter zijn dan de tijd die PAUSE aangeeft. Afhankelijk van de toepassing kan de PAUSE-tijd worden aangepast.

Als we echter de techniek uit het voorafgaande programma toepassen en voor elke tijdregistratie een aparte schakelaar gebruiken, is dit probleem niet aanwezig.

14 De tijd grafisch weergegeven

Het volgende programma doet tijdmetingen op een manier die overeenkomt met de methode uit hoofdstuk 13. Er is echter een interessant deel aan toegevoegd. De metingen worden opgeslagen in een ARRAY, een getallenrij die na de meting in een grafiek wordt weergegeven. Het programma laat zich het eenvoudigst demonstreren door een aantal malen (maximaal 10) met verschillende tijdsintervallen op de vuurknop te drukken.

```
1  REM de tijd grafisch weergegeven
10 PAPER 7: BORDER 7: INK 0
20 GOTO 100

30 REM subroutine tijdmeting
40 DEF FN s()=(PEEK 23672+256*PEEK 23673+65536*PEEK
  23674)/50
50 DEF FN g(a,b)=(a+b+ABS (a-b))/2
60 DEF FN t()=FN g (FN s(),FN s())
70 LET tijd=FN t()
80 RETURN

100 REM hoofdprogramma
110 LET n=1
120 DIM t(10)
130 LET t(1)=0
140 PRINT "TIJDMETING"
150 FOR i=1 TO 32: PRINT INK 5; "□";:NEXT i
160 INPUT "AANTAL METINGEN ";m

170 LET p=IN 31
180 IF p=255 THEN GOTO 170
190 POKE 23672,0: POKE 23673,0: POKE 23674,0
200 PAUSE 5
210 PRINT AT 0,20; INK 2; FLASH 1;" klok loopt "
220 PRINT AT 2,0
```


```

230 LET p=IN 31
240 IF p=255 THEN GOTO 230
250 LET n=n+1
260 IF n=m+1 THEN GOTO 1000
270 GOSUB 30
280 LET tijd=FN t()
290 LET t(n)=INT (tijd*10)/10
300 PAUSE 5
310 PRINT n-1,t(n)
320 GOTO 230

1000 REM uitvoer
1010 CLS: BORDER 1
1020 PRINT AT 3,4;"TIJD"
1030 PRINT AT 18,25;"WEG"
1040 PLOT 20,20: DRAW 0,135
1050 PLOT 20,20: DRAW 205,0
1060 PLOT 20,20
1070 FOR i=2 TO m
1080 LET interval=(t(i)-t(i-1))*10
1090 DRAW 20,interval
1100 NEXT i

```

Om het programma uit te proberen is de joy-stick een goed hulpmiddel. Geef als gewenst aantal meetpunten 10 op. In dit programma is dat het maximale aantal punten. Naar wens kan dit worden veranderd door regel 120 aan te passen. Druk kort op de vuurknop (korter dan PAUSE 5 zoals bij de stopwatch is beschreven). De klok loopt nu. Druk nu vijfmaal met een tussentijd van ongeveer een halve




Afbeelding 14.1 De grafiek zoals de computer die weergeeft.

seconde en dan vijfmaal zeer snel achter elkaar (bijvoorbeeld door de vuurknop gewoon vast te houden). De computer geeft dan de tijden in een grafiek weer zoals afbeelding 14.1 laat zien.

Het programma is te gebruiken voor allerlei natuurkunde-experimenten.

U herinnert zich misschien uit de natuurkundelessen nog het hellend vlak. Mocht dat niet zo zijn dan is uit ervaring wellicht bekend dat, wanneer een hellend vlak wordt betreden, het steeds sneller bergafwaarts gaat.



Afbeelding 14.2 Het nabootsen van het hellende vlak.

Voor het volgende experiment hebben we drie latjes, wat aluminiumfolie en een metalen kogel nodig. Uw garage heeft er misschien een, probeer het anders eens in het knikkerzakje van een buurkind. Afbeelding 14.2 laat zien wat de bedoeling is. De latten worden tegen elkaar gelijmd, gespijkerd of geschroefd en op elke 20 centimeter wordt met een punaise een strookje aluminiumfolie op de rails bevestigd. Aan beide zijden van de baan worden alle punaises met elkaar verbonden en met een draad met het massapunt en één van de ingangslijnen van de interfaceprint verbonden. De rails worden hellend opgesteld, de kogel wordt iets boven de eerste foliedetector gelegd en losgelaten. Omdat de snelheid van de kogel toeneemt, zal de tijd die nodig is om de opeenvolgende trajecten tussen de detectoren af te leggen steeds kleiner worden. Een prachtige opstelling dus om de theorie van het hellend vlak aan de praktijk te toetsen. In een programma kan dan eerst de benodigde informatie worden opgevraagd. Lengte van de intervallen, aantal intervallen, baanhelling enz. Bovendien kan in de grafiek ook de theoretische lijn worden getekend. Een mooie klus voor de natuurkundeleraars onder u. Een dergelijk programma kan ook worden aangepast voor metingen aan vallende voorwerpen. Voor dit doel beschrijft het volgende hoofdstuk een optische detector.

15 Optisch detecteren

Niet alleen voor natuurkunde-experimenten is een optische schakelaar handig. Detectie van bewegende voorwerpen is vaak gewenst, vooral als die voorwerpen toevallig mensen zijn. Het tellen van het aantal mensen in een ruimte kan met twee bij de ingang achter elkaar gemonteerde optische detectoren. De bewegingsrichting kan dan worden vastgesteld door de computer uit te laten zoeken welke detector het eerst een lichtstraalonderbreking meldt.

Zo'n richtinggevoelige lichtsluis kan ook bij inbraakbeveiliging worden toegepast. Of wat dacht u van een automatische klok voor de duivensport. Door de computer ook op dit gebied in te zetten, gaat er heel wat minder tijd verloren die dan op een andere wijze echt nuttig kan worden besteed.


Een optische schakelaar hoeft slechts uit enkele onderdelen te bestaan. Zoals het schema weergeeft, is een lichtgevoelige weerstand (LDR) in een spanningsdeler opgenomen. Deze is zo gedimensioneerd dat de poortuitgang hoog is als de LDR wordt belicht. Komt er echter een voorwerp tussen de lichtbron en de LDR dan stijgt de weerstandswaarde van de LDR en de poortuitgang wordt laag. Eén ingangskanaal wordt rechtstreeks met de poortuitgang verbonden. Bij toepassing van meer optische schakelaars die dezelfde functie hebben, is de programmatuur voor het uitlezen van de schakelaars wel heel eenvoudig:

```
10 LET p=IN 31
20 IF p<>255 THEN GOTO 10
30 PRINT"Er is een lichtstraal onderbroken"
40 GOTO 10
```

Het kan gewenst zijn de schakelaars afzonderlijk uit te lezen als we tevens informatie willen hebben over de plaats van de schakelaar die is geactiveerd. In dat geval krijgt elke detector in het programma zijn eigen getal p.

Naar wens kan de werking van de schakeling worden omgekeerd door de weerstand en de LDR van plaats te laten wisselen.


Om teleurstellingen bij verder experimenteren te voorkomen nog het volgende: Het gebruikte IC is een CMOS-schakeling van een type dat nagenoeg geen uitgangsstroom kan leveren. In ieder geval te weinig om er rechtstreeks een transistor mee te sturen. In hoofdstuk 28 wordt een hoogohmig transistortrapje beschreven dat wél door de hier toegepaste poortjes kan worden uitgestuurd.



Afbeelding 15.1 Optisch detecteren.

16 Schakelaar met instelbaar triggerniveau


Voor veel toepassingen is het detecteren van een aan/uit-conditie niet toereikend. Zo zal bij een temperatuurbeveiliging het niveau instelbaar moeten zijn. Het schema geeft een universele oplossing. In feite wordt de schakeling uit hoofdstuk 15 enigszins uitgebreid. De opnemer (temperatuur- of lichtgevoelige weerstand) is in een variabele spanningsdeler opgenomen, zie afbeelding 16.1. Het spanningsniveau op het knooppunt is instelbaar.



Afbeelding 16.1 De optische detector met spanningsdeler.

Boven een bepaalde spanning wordt de output van de poort laag. Met deze output sturen we de interface-ingang.

In hoofdstuk 15 is al aangegeven dat de passieve en de actieve weerstand kunnen worden verwisseld, zie ook afbeelding 16.2. Bij het opzetten van een printje kan hiermee rekening worden gehouden.



Afbeelding 16.2
De complete schakeling.

De veeleisende lezer zal een schakeling met vier kanalen op prijs stellen (er zitten immers vier poortjes in een 4093). Als op een door u te ontwerpen printje stekkerbusjes worden gesoldeerd, kan de print op de ingangspennen van de interface worden geschoven. De gevoelige elementen worden dan met een soepel snoertje met de print verbonden.

Deze schakeling kan ook goed als vensterindicator worden toegepast. Een vensterindicator bepaalt of een spanning tussen twee vooraf bepaalde grenzen ligt. Een mooi voorbeeld is de temperatuur in een aquarium of het niveau in een vloeistofreservoir.

Een programma voor de viervoudige detector heeft veel overeenkomst met de joy-stick-programma's. Na het reeds behandelde moet u uitstekend in staat zijn deze schakeling zelf te bouwen en vanuit een toepassingsprogramma uit te lezen.

17 Sinusgenerator met faseverschuiving

Zonder enige toevoeging van hardware kan de met de interface uitgebreide Spectrum als generator van complexe functies worden ingezet.

De sinusvorm speelt een belangrijke rol in de techniek. Hij is dan ook niet voor niets als standaardfunctie boven een van de Spectrum-toetsen te vinden. De analoge uitgang maakt uitvoer van die sinusfunctie als elektrisch signaal zonder meer mogelijk.

In het hier weergegeven programma kunnen twee sinussen met een onderlinge faseverschuiving worden gemengd. Voeren we een faseverschuiving 0 in, dan ontstaat er een ongestoorde sinus. Het aantal perioden van het signaal kan vrij worden gekozen. Doordat het programma, zoals alle programma's in dit boek, geen gebruik maakt van machinetaal kunnen er alleen laagfrequente signalen worden opgewekt. Maar dan ook extreem laagfrequent.

```
1 REM sinus met faseverschuiving
10 CLS: PAPER 7: BORDER 7: INK 0
20 PRINT " SINUSGENERATOR":PRINT
30 FOR i=1 TO 31: PRINT INK 5;"■";: NEXT i
40 OUT 31,0

50 LET f=0: LET x=0: LET n=0: LET y=0: LET s=0
60 LET p=IN 31
70 LET f=f+(p=247)-(p=251)
80 LET n=n+(p=254)-(p=253)
90 IF n<=0 THEN LET n=1
100 PRINT AT 3,4;"frequentie ";n;" Hz "
110 PRINT AT 4,4;"faseverschuiving ";f;" graden"
120 IF p<>239 THEN GOTO 60

130 FOR a=0 TO 360 STEP n
140 LET y=SIN (a*2*PI/360)+SIN ((a+f)*2*PI/360)
150 LET y=(400+y*127)/8
160 LET t=a*255/360
```

```


170 PLOT t,y
180 NEXT a

190 PLOT 0,50
200 DRAW 255,0
210 PRINT AT 8,0; INK 2;"AANTAL PERIODEN"
220 LET p=IN 31
230 LET s=s+(p=254)-(p=253)
240 IF s<1 THEN LET s=1
250 PRINT AT 8,16;s;" "
260 IF p<>239 THEN GOTO 220

270 FOR b=1 TO s
280 FOR a=0 TO 360 STEP n
290 LET y=127+127*(SIN (a*2*PI/360)+SIN
    ((a+f)*2*PI/360))/2
300 OUT 31,y
310 PRINT AT 9,19;Y
320 NEXT a
330 PRINT AT 8,19;B;" "
340 NEXT b
350 OUT 31,0

```

In het blok worden vanaf regel 50 de frequentie en de faseverschuiving vastgelegd. Het deel vanaf regel 130 laat de op te wekken golfvorm zien. In het blok dat bij 190 begint, kan het aantal uit te voeren perioden worden opgegeven terwijl het laatste blok voor de verzending zorgdraagt.



Afbeelding 17.1 Zo ziet de sinus er uit.

Regel 290 verdient wat meer aandacht. De spanning op de analoge uitgang kan niet negatief worden. We moeten er daarom voor zorgen dat het niveau waarop de sinus zich bevindt op de helft van de maximale uitgangsspanning ligt. De uitgang is maximaal bij het getal 255. De halve uitgangsspanning ligt dan bij 127. Door de y-waarde van regel 290 met een tussen 0 en 1 instelbaar getal te vermenigvuldigen, kan er een volumeregeling aan worden toegevoegd. Het naar buiten brengen van de golfvorm gebeurt in regel 300 met de eenvoudige OUT-opdracht. De in het programma opgegeven frequentie zal niet overeenkomen met de werkelijke frequentie. De verkregen tijden worden immers enigszins beïnvloed door het programma zelf.

Dat er op dit programma vele varianten mogelijk zijn, laat het volgende hoofdstuk zien.

18 Dubbelsinusgenerator

Het voorgaande programma heeft de zeer veelzijdige mogelijkheden van de Spectrum als functiegenerator al gedemonstreerd. Als twee sinussen met verschillende frequentie worden gemengd, ontstaat er interferentie. We kunnen dit eenvoudig in een programma simuleren. Het effect kan eerst op het scherm worden beoordeeld en vervolgens als echt elektrisch signaal op de analoge uitgang verschijnen. Ook hier is het aantal perioden vrij te kiezen.

```
1 REM dubbelsinus
10 CLS: PAPER 7: BORDER 7: INK 0
20 PRINT " DUBBELSINUS GENERATOR":PRINT
30 FOR i=1 TO 31: PRINT INK 5; "■";:NEXT i
40 OUT 31,0
50 LET f1=0: LET f2=0: LET x=0: LET y=0
60 LET s=0: LET b=0
70 LET p=IN 31
80 LET f2=f2+(p=247)-(p=251)
90 LET f1=f1+(p=254)-(p=253)
100 IF f1<=0 THEN LET f1=1
110 IF f2<=0 THEN LET f2=1
120 PRINT AT 3,4;"frequentie 1 ";f1;" Hz"
130 PRINT AT 4,4;"frequentie 2 ";f2;" Hz"
140 IF p<>239 THEN GOTO 70

150 FOR a=0 TO 360 STEP f1
160 LET b=b+f2
170 LET y=SIN (a*2*PI/360)+SIN (b*2*PI/360)
180 LET y=(400+y*127)/8
190 LET t=a*255/360
200 PLOT t,y
210 NEXT a
220 PLOT 0,50: DRAW 255,0
```




```

230 PRINT AT 8,0; INK 2;"AANTAL PERIODEN"
240 LET p=IN 31
250 LET s=s+(p=254)-(p=253)
260 IF s<1 THEN LET s=1
270 PRINT AT 8,16;s;" "
280 IF p<>239 THEN GOTO 240

290 LET b=0
300 FOR c=1 TO s
310 FOR a=0 TO 360 STEP f1
320 LET b=b+f2
330 LET y=127+127*(SIN (a*2*PI/360)+SIN
    (b*2*PI/360))/2
340 OUT 31,y
350 PRINT AT 9,19;y
360 NEXT a
370 PRINT AT 8,19;c;" "
380 NEXT c
390 OUT 31,0

```

De spaties in de regels 270 en 370 dienen weer om restanten van voorafgaande getallen te wissen. Het eerste blok vraagt de gegevens op. Deze worden ingevoerd met de joy-stick. De frequenties worden ingesteld met een horizontale en een verticale beweging. Als de juiste waarde is verkregen, kan de vuurknop worden ingedrukt. De functie wordt dan op het scherm afgebeeld. Het aantal uit te voeren perioden kan weer met de joy-stick worden ingesteld. Een nieuwe druk op de vuurknop en de analoge uitgang stuurt het signaal de wereld in.



Afbeelding 18.1 De dubbele sinus.

Met de hier gegeven programma's als voorbeeld moet het mogelijk zijn elke gewenste golfvorm te genereren. De wiskundigen onder u weten vast wel toepassingen te verzinnen. Bijvoorbeeld het onderzoek naar convergentie of divergentie van functies of het vaststellen van de herhalingsfrequentie van periodieke functies.

19 Een karaktervolle lichtpen

De lichtpen is net als het toetsenbord, de joy-stick en de muis een uiterst nuttig hulpmiddel bij de conversatie met de computer.

Door de lichtpen kan de computer ook gebarentaal begrijpen. Aanwijzen is voldoende om de meest complexe processen tot stand te brengen.

Een lichtpen kan op verschillende niveaus werkzaam zijn. Het is denkbaar dat hij op pixelniveau werkt, dus met een hoog oplossend vermogen. Ook kan hij op karakterniveau werken. Het karakter is dan het kleinste onderdeel dat kan worden onderscheiden. Op welk niveau de pen ook dient te werken, enige kennis van de beeldopbouw is noodzakelijk.

TV-beeldopbouw

Een TV-beeld wordt opgebouwd door een lichtpuntje met grote snelheid langs een lijnenpatroon te voeren. Dit lijnenpatroon wordt 50 maal per seconde doorlopen zodat ons oog een vast beeld waarneemt. Richtten we nu een lichtgevoelig element op de beeldbuis, dan zal er, steeds wanneer de lichtstip voorbijkomt, een pulsje in de lichtopnemer worden opgewekt. Een rekensommetje kan ons dan in principe vertellen welk adres in het videogeheugen op dat ogenblik werd uitgelezen. Vervolgens kunnen we dan naar die geheugenplaats informatie sturen. Bijvoorbeeld PLOT een oplichtende punt of CIRCLE met de desbetreffende coördinaten. Het bovenstaande zou eenvoudig zijn als de ordening van het videogeheugen overeen zou komen met de TV-beeldopbouw. Niets is echter minder waar.

Het videogeheugen

In het handboek kunnen we vinden dat het videobeeld is opgeslagen tussen de adressen 16384 en 22528. In tegenstelling tot sommige andere geheugenfuncties liggen deze adressen vast. Op elk adres zijn acht bits opgeslagen waarmee één lijn van een karakter is vastgelegd. Elk karakter is opgebouwd uit acht van die lijnen. De decimale waarde van deze acht bits ligt dus tussen 0 en 255.

1	2	4	8	16	32	64	128
---	---	---	---	----	----	----	-----

<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
--------------------------	--------------------------	--------------------------	--------------------------	--------------------------	--------------------------	--------------------------	--------------------------

Zo kunnen we het bovenste lijntje van het eerste karakter zelf invullen met de opdracht

POKE 16384,25

We zien dan  linksboven op het scherm.

Het zou aardig zijn om alle mogelijke pixels één voor één te laten oplichten. Zodra zich dan een oplichtend pixel onder de lichtpen bevindt, kunnen we de desbetreffende geheugenlocatie opvragen.

Het volgende voorbeeld zal echter duidelijk maken dat een dergelijke aanpak in BASIC niet erg geslaagd is. Het programma laat de pixels stuk voor stuk oplichten maar vraagt hiervoor een zee van tijd.

```
1 REM pixels
10 BORDER 0: PAPER 0: INK 7
20 DIM i$(704)
30 PRINT OVER 1;i$
40 FOR i=16384 TO 22528
50 FOR j=7 TO 0 STEP -1
60 POKE i,2^j
70 NEXT j
80 POKE i,0
90 NEXT i
```

Het grootste deel van deze tijd wordt ook hier weer opgeslokt door de berekening in regel 60.

Door naar alle geheugenplaatsen het getal 255 te schrijven, is het volgen van de beeldopbouw erg mooi te zien. Bovendien wordt de machtsverheffing op deze manier omzeild.


```
1 REM karakterlijnen
10 FOR i= 16384 TO 22528
20 POKE i,255
30 NEXT i
```

Om het leven niet moeilijker te maken dan het al is, beperken we ons bij het lichtpenontwerp tot het karakterniveau.

We laten een karakterpositie oplichten en detecteren met de lichtpen op welke plaats deze zich bevindt.

De hardware

Het elektronisch schema in afbeelding 19.1 is, dank zij digitale techniek, weinig gecompliceerd.



Afbeelding 19.1 Het schema van de lichtopnemer.

Als lichtopnemer wordt een lichtgevoelige weerstand gebruikt. Deze is opgenomen in een weerstanddeler die met de instelpotentiometer kan worden afgeregeld. Het niveau dat op het knooppunt van de deler aanwezig is wordt door een NAND-poort verder verwerkt.

Het bijzondere van deze poort schuilt in de triggerwerking. Als de spanning boven het aanspreekniveau komt, wordt de uitgang laag. Om weer hoog te worden, moet de aangeboden spanning enigszins beneden de aanspreekspanning worden gebracht. Hierdoor is er een dood gebiedje tussen de schakelniveaus, waardoor een storingsongevoelige werking wordt verkregen. De tweede poort wordt gebruikt om het signaal te inverteren. De voeding wordt uit de 5 volt-aansluiting van de interfaceprint betrokken. De signaaldraad wordt met een willekeurige ingang verbonden. In rust is de uitgang hoog zodat de interface het getal 255 geeft. Dat heeft als voordeel dat de pen (evenals de joy-stick) normaal aangesloten kan blijven als hij niet in gebruik is.

Er mag dan natuurlijk geen omgevingslicht op de lichtgevoelige weerstand vallen. Een beschermkapje kan daarvoor zorgen. Om het stroomverbruik nog verder te drukken (de lichtpen neemt maar weinig op) worden de niet gebruikte ingangen aan massa gelegd.

Hoewel de weinige onderdelen gemakkelijk op een stukje gaatjesprint passen, is toch het printontwerpje opgenomen. De signaalleiding wordt op één van de ingangspennen geprikt. Neem voor de LDR een courant exemplaar en scherm het geheel af voor omgevingslicht. Afbeelding 19.2 laat zien dat de lichtpen een geduchte concurrent voor de 'muis' kan zijn.

Het uitproberen van de lichtpen kan uitstekend gebeuren met het volgende programma. Er verschijnt een sterretje op de plaats waar de lichtvlek de pen passeert.

```
1 REM lichtpen
10 BORDER 0
```

```

20 DIM i$(704)
30 PRINT AT 0,0; OVER 1; PAPER 0;i$
40 PLOT 0,0
50 DRAW INK 6;255,0
60 DRAW INK 6;0,175
70 DRAW INK 6;-255,0
80 DRAW INK 6;0,-175
90 FOR j=0 TO 21
100 FOR i=0 TO 31
110 PRINT AT j,i; OVER 1; BRIGHT 1;"[]"
120 IF IN 31=255 THEN PRINT AT j,i; INVERSE 1;"*"
130 PRINT AT j,i; OVER 1;"■"
140 NEXT i
150 NEXT j
160 GOTO 10

```

Het hier beschreven programma lijkt erg traag voor algemene toepassingen. In hoofdstuk 20 zullen we zien dat de snelheid drastisch kan worden opgevoerd.



Afbeelding 19.2 De licht-'muis'-pen.

20 De pen als dirigeerstok

Met de BEEP-opdracht heeft de Spectrum de mogelijkheid om zich muzikaal te uiten. De opdrachten moeten echter in getalvorm worden gegeven en het componeren van leuke liedjes is dan ook een hele klus. De notenbalk is een handig hulpmiddel bij het maken van muziek. We gebruiken de lichtpen om een muziekstukje op de notenbalk te creëren.

```
1 REM muziek
10 LET h=1: LET kolom=3
20 DIM t(30)
30 DIM i$(704)
40 BORDER 1
50 PRINT AT 0,0; OVER 1; PAPER 1;i$
60 FOR i=1 TO 9
70 PLOT INK 7;0,146-i*16
80 DRAW INK 7; BRIGHT (INT (i/2)<>i/2);255,0
90 NEXT i

100 FOR i=1 TO 9
110 PRINT AT 3+2*i,(INT (i/2)=i/2); BRIGHT 1; INK 7;
    "■ ■"
120 PAUSE 2
130 LET p=IN 31
140 IF p=255 THEN GOSUB 8000
150 IF kolom=31 THEN GOTO 200
160 PRINT AT 3+2*i,(INT (i/2)=i/2); INK 0;"■ ■"
170 NEXT i
180 GOTO 100

200 FOR h=3 TO 30
210 BEEP .2,t(h)
220 NEXT h
230 GOTO 200
```




```
8000 REM subroutine noten
8010 LET kolom=kolom+1
8020 PRINT AT 3+2*i,kolom; PAPER 1; INK 6; FLASH 1;"0"
8030 BEEP .5,56-2*i
8040 LET h=h+1
8050 LET t(h)=56-2*i
8060 RETURN
```

Het eerste blok van dit programma tekent de notenbalk. Door de uitdrukking achter BRIGHT worden de lijnen afwisselend normaal en extra helder getrokken. Het tweede blok zorgt voor oplichtende markeringen waar de lichtpen op kan reageren. Als de lichtpen op zo'n flitsend punt wordt gericht, komt de subroutine in werking. Afhankelijk van de plaats op de notenbalk wordt een toonhoogte vastgelegd. De toonhoogte $t(h)$ wordt hier bepaald door de formule $56-2*i$. Om zuivere tonen te maken, zult u hier iets anders voor moeten bedenken. Er blijft dus voldoende zelfwerkzaamheid over. Zo kan dit programma door de echte liefhebber worden uitgebreid met kruisen en mollen, kwart maten en andere bij de muzieknotatie gebruikelijke handigheidjes.

21 Lichtnetschakelaar

Door een uitgangslijn van de Spectrum-interface met een triac te verbinden kunnen we lampen en motoren schakelen. Afbeelding 21.1 toont deze **levensgevaarlijke** oplossing.



Afbeelding 21.1 Een levensgevaarlijke oplossing voor het sturen van lampen!



De Spectrum-print komt op deze wijze aan het net te hangen en via de draadverbindingen ook alle randapparatuur. Bij het schakelen van netfuncties is een **volledige elektrische isolatie** noodzakelijk. Laat u dan ook nooit verleiden tot experimenten met een directe koppeling.

Met wat extra onderdelen is een veilige verbinding tot stand te brengen, zie afbeelding 21.2. Hoe dat in zijn werk gaat wordt in dit hoofdstuk beschreven.


De interface-uitgangslijn laat een lichtgevende diode oplichten. De weerstand van 330 ohm dient hierbij als stroombegrenzing. Als de LED oplicht, wordt de lichtgevoelige transistor opengestuurd. Deze verzorgt op zijn beurt de ontsteekstroom voor de triac. Op deze wijze kan een transistor de triac opensturen zowel tijdens de positieve als de negatieve periode van de netspanning.

De combinatie LED-fototransistor is opgenomen in een enkel onderdeel.

Om een volledige scheiding met de netspanning te realiseren, mag de lichtgevoelige stuurtransistor niet vanuit de Spectrum worden gevoed. Voor de voe-



Afbeelding 21.2 Het schema voor de TIL 111.



Afbeelding 21.3 De geheel gemonteerde lichtnetschakelaar aangesloten op de interface.

ding gebruiken we een schakelingetje dat 12 volt maakt van de netspanning. Stabilisatie met een zenerdiode voldoet ruimschoots aan de te stellen eisen.

Om de mogelijkheden van deze optische koppeling aan netapparatuur volledig te kunnen benutten, is er een printontwerp achterin dit boek opgenomen waarop ruimte is voor vier kanalen.

Let op! Hoewel de interface (en daarmee de computer) veilig van het net is gescheiden, voert de schakelprint uiteraard wel gevaarlijke spanningen. Het is dan ook van groot belang dat de print in een kunststof-behuizing wordt opgenomen. Om toch een gemakkelijke koppeling met de interfaceprint te bewerkstelligen, kan de verbinding met een stukje 'flat-cable' tot stand worden gebracht. Voor dit doel is een stekkerprintje als transfer opgenomen.

Vanuit een BASIC-programma kunnen met eenvoudige OUT 31,n-opdrachten alle mogelijke combinaties van de vier kanalen worden geschakeld. We denken hierbij aan oplichtende discovloeren, looplichten, lichtorgels, reclameverlichting en wat er zoal nog meer op dit gebied is te verzinnen.

Maar ook een automatisch pompstation of de bediening van een aquarium, centrale verwarming of een kweekkas behoort tot de mogelijkheden.

Het testen van de print kan gebeuren met het programma uit hoofdstuk 5. Let erop dat programma's voor deze schakeleenheid altijd dienen te beginnen met

OUT 31,0

om met een schone (uitgangs)lei te kunnen starten.

22 Toneelverlichting met de pen

Grote delen van de moderne procesindustrie spelen zich op beeldschermen af. Bent u op enige wijze betrokken bij een (amateur)toneelvereniging? Breng dan een stukje van die moderne techniek over. In dit hoofdstuk brengen we het toneel met alle aanwezige lichtbronnen in kaart. Als voorbeeld nemen we een toneel met elektrisch bediende gordijnen, drie spotlights en vier groepen gekleurd effectlicht. Door met de lichtpen een lamp op het scherm aan te wijzen, wordt deze aan- of uitgeschakeld.

```
1 REM toneelverlichting
10 DIM a(8): DIM u(8)
20 LET uit=0
30 OUT 31,0
40 GOSUB 8000
50 GOTO 1000

100 REM subroutine cursorlicht
110 PRINT AT r,k; BRIGHT 1; INK 7; "■ ■"
120 PAUSE 2
130 LET p=IN 31
140 IF p=255 THEN LET a(1)=ABS (a(1)-1)
150 IF p=255 THEN PRINT AT 1,1; PAPER 1; INK 7;
    INVERSE 1; "*"
160 PRINT AT r,k; INK 0; "■ ■"
170 PAUSE 2
180 RETURN

200 REM subroutine plaatsbepaling
210 IF a(1) THEN PRINT AT r-1,k; INK 6; BRIGHT 1;
    PAPER 1; "AAN"
220 IF NOT a(1) THEN PRINT AT r-1,k; INK 6; BRIGHT 1;
    PAPER 1; "[ ] [ ] [ ]"
230 GOSUB 100
240 RETURN
```

```

1000 LET l=1: LET r=4: LET k=7
1010 GOSUB 200
1020 LET l=2: LET r=4: LET k=23
1030 GOSUB 200
1040 LET l=3: LET r=10: LET k=15
1050 GOSUB 200
1060 LET l=4: LET r=21: LET k=7
1070 GOSUB 200
1080 LET l=5: LET r=21: LET k=12
1090 GOSUB 200
1100 LET l=6: LET r=21: LET k=17
1110 GOSUB 200
1120 LET l=7: LET r=21: LET k=22
1130 GOSUB 200
1140 LET l=8: LET r=13: LET k=0
1150 IF a(1) THEN PRINT AT r-1,k; INK 6; BRIGHT 1;
      PAPER 1;"open "
1160 IF NOT a(1) THEN PRINT AT r-1,k; INK 1;"dicht"
1170 GOSUB 100

1200 FOR u=1 TO 8
1210 LET uit=uit+a(u)*2^(u-1)
1220 NEXT u
1230 PRINT AT 0,3;uit;" "
1240 OUT 31,uit

1250 LET uit=0
1260 IF INKEY$<>" THEN STOP
1270 PRINT AT 1,1; INK 1;"["
1280 GOTO 1000

8000 DIMa$(704)
8010 PRINT AT 0,0; OVER 1; PAPER 1;a$
8020 BORDER 1: INK 7
8030 PLOT 0,80: DRAW 50,0
8040 PLOT 205,80: DRAW 50,0: DRAW 0,95: DRAW -255,0:
      DRAW 0,-95
8050 INK 5
8060 PLOT 0,50: DRAW 255,0
8070 INK 0
8080 CIRCLE 70,140,20: CIRCLE 180,130,30: CIRCLE
      120,100,60
8090 PRINT AT 19,5; INK 2;"■■■■■"

```

```

8100 PRINT AT 19,10; INK 4;"■■■■■"
8110 PRINT AT 19,15; INK 6;"■■■■■"
8120 PRINT AT 19,20; INK 3;"■■■■■"
8130 PRINT AT 13,0; PAPER 1; INK 5;"■ ■ ■ ■ ■"
8140 PRINT AT 13,26; PAPER 1; INK 5;"■ ■ ■ ■ ■"
8150 PRINT AT 16,22; PAPER 1; INK 7;"ENTER=stop"
8160 RETURN

```

Het programma is eenvoudig te doorgronden. De subroutine vanaf 8000 tekent het podium met de lichtgroepen. In het korte hoofdprogramma vanaf regel 1000 wordt de plaats van de oplichtende cursor vastgelegd. Dit oplichten gebeurt in het blok dat op regel 110 start. De subroutine vanaf regel 200 laat op het scherm zien welke lampen aan of uit zijn. Dit programma geeft een goede indruk van de mogelijkheden van de computer bij het verrichten van complexe taken.


23 Analoge ingang

De voorafgaande hoofdstukken hebben laten zien dat de binaire ingang van de interface legio mogelijkheden biedt tot koppeling van de Spectrum met de buitenwereld. Een analoge ingang die veranderende spanningen kan meten geeft nog meer mogelijkheden.


In dit hoofdstuk beschrijven we een schakeling die op de inmiddels bekende manier met een IN 31-opdracht wordt uitgelezen. De schakeling is vooral bedoeld voor het bewaken van signalen.

We maken een nogal merkwaardig gebruik van een IC dat eigenlijk is bedoeld om een rij LED's aan te sturen. De normaal gebruikelijke schakeling rondom dit IC is in afbeelding 23.2 afgebeeld.

De ingang bestaat uit een spanningsdeler. Tussen de weerstanden zijn differentiaalversterkers opgenomen die de aangeboden ingangsspanning vergelijken met de spanning op de spanningsdeler. Als de signaalspanning de delerspanning overschrijdt, klapt de uitgang van de differentiaalversterker om en de bijbehorende LED gaat branden.



Afbeelding 23.1 De UAA 170(L).




Afbeelding 23.2 Het aansluiten van de UAA 170(L).

De LEDjes zijn in vier groepen van vier aangesloten. De stroom loopt hierbij vanuit één van de transistoren met aansluitingen 6, 7, 8 en 9 naar één van de transistoren die verbonden zijn met 2, 3, 4 en 5. Via één van de eerstgenoemde transistoren wordt een LED met de plus verbonden, terwijl de laatstgenoemde transistoren verbinding geven met een stroombron die de LED-stroom constant houdt. Door van elke groep van vier transistoren er één open te sturen, kan elk van de zestien LED's afzonderlijk oplichten.

Door de eigenaardige aansluitwijze van de LED's zijn er niet direct zestien kanalen beschikbaar die of hoog of laag zijn. Bovendien hebben we op de interface geen zestien ingangen. We moeten de uitgangen dan ook decoderen op een manier die voor de interface begrijpelijk is.

De attente lezer zal het verband tussen de tweemaal vier IC-aansluitingen en de acht interface-ingangslijnen al hebben gelegd. We koppelen de acht LED-uitgangen op de een of andere wijze aan de acht interface-uitgangen en laten de com-



Afbeelding 23.3 De emitter naar massa of de collector naar plus.


puter uitzoeken welk getal bij een bepaalde combinatie hoort. Alleen voor de koppeling moeten we dan een schakeling verzinnen. Dit wordt ons echter door de ontwerper van het IC nogal moeilijk gemaakt. Als voorbeeld nemen we één van de uitgangen 6 t/m 9, zie afbeelding 23.3. Het lijkt voor de hand liggend vanaf de emitter een weerstand naar massa te leggen.

Het knooppunt is dan laag als de transistor niet is opengestuurd en hoog als de transistor wel wordt opengestuurd. Het enige probleem zou dan nog het te hoge niveau zijn. De schakeling geeft een spanning van 9 volt af, terwijl de interface niet meer dan 5 volt accepteert. Afgezien hiervan zou voor de transistoren 2 t/m 5 een weerstand naar de plus het probleem op kunnen lossen.

De interne elektronica van de UAA 170 is echter zodanig opgebouwd dat dit verhaal niet opgaat. Het is een vereiste dat de stroom niet door weerstanden wordt belemmerd (er zit immers een stroombron in de UAA 170) en dat de vier transistoren uit een groep door identieke LEDs worden belast. Dit laatste geldt ook voor de bovenste vier transistoren.

Alle problemen worden in één klap opgelost met een handjevol 'optocouplers'. Zo'n optische koppeling bevat een LED en een lichtgevoelige transistor. Als de LED brandt, wordt de transistor opengestuurd.

In afbeelding 23.4 is een en ander uitgewerkt.



Afbeelding 23.4 Het aansluiten van de LEDs, let op de stroombegrenzing.

Let erop dat voor de LEDs 6 t/m 9 een stroombegrenzingsweerstand is opgenomen, terwijl de LEDs 2 t/m 5 gebruik maken van de ingebouwde stroombron. Aan de transistorkant van de koppelingen leggen we nu zonder problemen alle emitters aan de voedingsnul. De optrekweerstand op de interfaceprint zorgen er dan voor dat de ingang in rust hoog is, terwijl de opto-transistor het signaal naar massa trekt als hij door de LED wordt belicht. De verschillende spanningsniveaus tussen A-D-omzetter en interface worden op deze wijze mooi omzeild.

Nog een opmerking ten aanzien van de stroombron:


De stroom is in te stellen via de aansluitingen 15 en 16. Met een paar vaste weerstanden kan de voor de optocouplers vereiste waarde worden vastgelegd.

De ingangsschakeling

De UAA 170 is voorzien van ingangsaansluitingen die gecombineerde functies

hebben. Er zijn twee ingangen die het bereik van hetingangssignaal vastleggen. De spanning op punt 12 (V_{min}) bepaalt de spanning aan de onderzijde van de interne weerstandsdeler, terwijl punt 13 (V_{max}) hetzelfde doet voor de bovenkant van die spanningsdeler. Deze stuurspanningen mogen nooit hoger worden dan 6 volt omdat boven die waarde het IC stuk gaat.

Door de combinatie van functies worden V_{min} en V_{max} ook gebruikt om een meer of minder vloeiende lichtovergang tussen de LEDs in te stellen. Als de verschilspanning tussen V_{max} en V_{min} lager is dan 4 volt dan is de overgang tussen de LEDs vloeiend. Dat wil zeggen dat er twee of zelfs meer LEDs tegelijk kunnen branden. Is de verschilspanning hoger dan 4 volt, dan is de overgang abrupt. Het zal duidelijk zijn dat dit laatste gewenst is. We willen immers een eenduidige uitlezing hebben omdat de computer alleen 'helemaal uit' of 'helemaal aan' herkent. De aansluitingen V_{min} en V_{max} kunnen met een eenvoudig spanningsde-
lertje (afb. 23.5) op de juiste waarde worden gebracht.




Afbeelding 23.5 De spanningsdeler.

De juiste waarden voor de delerweerstand komen verderop aan de orde.

Een extra versterkertje

In veel gevallen zullen de signalen die we door de computer willen laten interpreteren niet precies tussen V_{min} en V_{max} liggen. Vooral de elektronicus is geïnte-



Afbeelding 23.6 De extra versterkertrap.


resseerd in kleinere signalen. Voordat de hiervoor beschreven schakeling dan ook op een printje wordt gezet, breiden we het geheel uit met een extra versterkertrapje.

Afbeelding 23.6 geeft de schakeling rondom enkele op-ampjes weer. Er zitten er vier in een huisje. We gebruiken er maar drie van. Misschien heeft u er moeite mee dat er één ongebruikt blijft. De prijs van het IC is echter zodanig dat de gewetensbezwaren niet erg groot hoeven zijn.

Omdat operationele versterkers gevoed dienen te worden met een dubbele voedingsspanning is rondom A1 een kunstmatige dubbele voeding gerealiseerd. Let wel, de kunstmatige massa ligt nu op 4,5 volt (als we uitgaan van een voedingsspanning van 9 volt). De eigenlijke versterker rondom A2 is instelbaar tussen 1 en 500 maal versterking. Omdat we een stabiele werking willen hebben over het hele versterkingsbereik, is de regeling uitgevoerd met een tienslagen-instelpotentiometer. Iets duurder, maar hier onmisbaar. Eventueel kan de weerstandswaarde van de potentiometer worden verhoogd voor nog grotere versterkingen. De versterking is vastgelegd met de formule die in het schema is gegeven. Opzettelijk is voor deze versterker een niet inverterende schakeling gekozen. De ingangsimpedantie is daardoor zo hoog dat een voorafgaande schakeling er niet noemenswaardig door wordt belast. Omdat de kunstmatige massa op halve voedingsspanning ligt, beweegt de uitgang van de versterker zich tussen de halve en de volledige voedingsspanning.

Als we dit aan de UAA 170 aanbieden, hebben we netjes voldaan aan de voorwaarde dat er minimaal 4 volt verschil in minimale en maximale uitsturing is. De spanningen overschrijden dan echter wel de maximaal toegestane 6 volt. Dit probleem is simpel opgelost met een tweede versterkertrapje rondom A3 dat het signaal inverteert. De versterking is precies 1x. De uitgang beweegt zich na het inverteren tussen halve voedingsspanning en massa.

Er is nu nog slechts een schoonheidsfoutje aanwezig. Omdat de door de Spectrum afgegeven ongestabiliseerde spanning in werkelijkheid hoger is dan 9 volt (ongeveer 12 volt), beweegt de versterkeruitgang zich tussen 5,96 en 1,29 volt bij een ingangsspanning tussen 0 en 4,1 volt. De versterkingsregeling is hierbij volledig teruggedraaid (tegen de klok in). Zoals de genoemde getallen aangeven, is de versterking dan ongeveer 1x. Voor een optimale aanpassing tussen versterker en UAA 170 kunnen nu de weerstanden in de spanningsdeler worden gekozen.



Afbeelding 23.7 Gedimensioneerde spanningsdeler.


Bij de gegeven weerstandswaarden liggen de spanningen V_{min} en V_{max} op 1,66 en 5,96 volt.

Dit bereik ligt binnen het versterkerbereik zodat alle indicatiemogelijkheden van de UAA170 doorlopen worden bij een oplopende meetwaarde.

Bij een versterkingsfactor van 1 is het ingangsspanningsbereik van de versterker 0-4 volt.

Bij te kleine signalen wordt de versterking verhoogd met behulp van de tienslageninstelpotentiometer. Bij grotere waarden dan 4 volt moet er een spanningsdeler (afb. 23.8) voor de ingang worden opgenomen. Deze mag hoogohmig zijn.


Afbeelding 23.8 Ingangsspanningsdeler.




In hoofdstuk 24 wordt de bouw van de A/D-omzetter onder handen genomen.

24 De bouw van de A/D-omzetter


Voor het maken van de print is de lay-out achterin het boek opgenomen. Het monteren van de onderdelen aan de hand van de afbeeldingen moet probleemloos kunnen verlopen. Let op de inkepingen van de IC's. Bij de optocouplers is pen nummer 1 aangegeven met een markering. De twee voedingseleco's dienen ook in de juiste stand te worden gemonteerd. Het solderen van de stekkerverbinding kan weer het beste op de interfaceprint gebeuren. Het is dan zeker dat de busjes goed uitgelijnd staan.



Afbeelding 24.1 Het schema van de A/D-omzetter.




Afbeelding 24.2 De componentenopstelling.




Afbeelding 24.3 De gemonteerde print.

De componentenopstelling in afbeelding 24-2 op bladzijde 77 kan verwarrend scheppen: de A/D-omzetter moet worden aangesloten op de ingangspennen van de interface-print. Het inschrift INTERFACE UIT duidt op de uitgang van de A/D-omzetter! (zie ook afbeelding 24-4.)

De nog niet besproken spanningsdeler rondom de verticaal gemonteerde instelpotentiometer is niet met de versterkingang doorverbonden. Deze ingang is dan ook zonder meer te gebruiken als voltmeter. Willen we echter een passief element (LDR of NTC) aansluiten, dan moeten de beide met IN gemerkte pennen worden doorverbonden. Daardoor ontstaat de in afbeelding 24.4 aangegeven situatie.




Afbeelding 24.4 Het aansluiten van de LDR.



Test

Om een goed begrip te krijgen van de werking van de A/D-omzetter sluiten we een (instel) potentiometer aan zoals in afbeelding 24.5 is aangegeven. We maken daarbij dus geen gebruik van de extra op de print gemonteerde potentiometer.




Afbeelding 24.5 Het aansluiten van de instelpotentiometer.

Door de loper te verdraaien, kunnen we de ingang op verschillende spanningen brengen. De versterkingsregeling wordt helemaal teruggedraaid (versterking 1x).

We typen nu het volgende programma in:

```
10 LET p=IN 31
20 PRINT AT 0,0;p;" "
30 GOTO 10
```

De spaties in regel 20 wissen eventuele restanten van een vorige meting uit. Als we nu de loper van de potentiometer langzaam van de massa-aansluiting naar de voedingsspanning verdraaien, zien we linksboven op het scherm de waarde afgedrukt die de interface bij de desbetreffende spanning ziet. De tabel geeft de rij getallen die achtereenvolgens verschijnen. De interfacekanalen die op dat ogenblik nul zijn, zijn met een kruisje aangegeven. Bovendien is voor een aantal punten de ingangsspanning genoteerd.



Afbeelding 24.6 16 lineaire uitleesniveaus.

Hoewel de verschilspanning tussen de UAA170-ingangen V_{min} en V_{max} groter dan 4 volt is, zien we dat er toch combinaties optreden waarbij er meer dan één LED uit een groep van vier oplicht. Omdat deingangsspanningen die bij deze combinaties horen erg dicht bij de waarden liggen waarbij maar één LED per groep oplicht, hebben we er niet veel aan. Alleen de combinaties waarbij de ingangsspanningen zijn opgegeven, hebben praktisch nut. We zien inderdaad 16 lineair verdeelde uitleesniveaus zoals door de grafiek in afbeelding 24.6 wordt geïllustreerd.

Getal	Interfacekanaal							Ingangs- spanning	
	3	2	1	0	4	5	6		7
119	X							X	0
55	X						X	X	
183	X						X		0,36
151	X					X	X		
215	X					X			0,63
199	X				X	X			
231	X				X				0,90
227	X	X		X					
235		X		X					1,19
203		X		X	X				
219		X				X			1,45
155		X				X	X		
187		X					X		1,71
59		X					X	X	
123		X						X	2,00
122		X		X				X	
126				X				X	2,26
62				X				X	
190				X			X		2,52
158				X		X	X		
222				X		X			2,79
206				X	X	X			
238				X	X				3,04
236			X	X	X				
237			X		X				3,31
205			X		X	X			
221			X			X			3,59
157			X			X	X		
189			X				X		3,84
61			X				X	X	
125			X					X	4,10

25 Voltmeter-software

Een voltmeterprogramma moet de rij interfacegetallen naar een spanning kunnen vertalen. We zullen een voltmeter bouwen met een volleschaalbereik van 1,6 volt.

De interfacetabel is dan vastgelegd:

Waarde	Spanning (volt)
119	0,1
183	0,2
215	0,3
231	0,4
235	0,5
219	0,6
187	0,7
123	0,8
126	0,9
190	1,0
222	1,1
238	1,2
237	1,3
221	1,4
189	1,5
123	1,6

Deze tabel kunnen we in een programma vastleggen.

```
10 LET p=IN 31
20 IF p=119 THEN LET v=0.1
30 IF p=183 THEN LET v=0.2
40 IF p=215 THEN LET v=0.3
50 IF p=231 THEN LET v=0.4
60 IF p=235 THEN LET v=0.5
70 IF p=219 THEN LET v=0.6
80 IF p=187 THEN LET v=0.7
```

```

90 IF p=123 THEN LET v=0.8
100 IF p=126 THEN LET v=0.9
110 IF p=190 THEN LET v=1.0
120 IF p=222 THEN LET v=1.1
130 IF p=238 THEN LET v=1.2
140 IF p=237 THEN LET v=1.3
150 IF p=221 THEN LET v=1.4
160 IF p=189 THEN LET v=1.5
170 IF p=125 THEN LET v=1.6
180 PRINT AT 0,0;"De spanning is ";v;" Volt"
190 GOTO 10

```

De voltmeter werkt nu. Sluit een spanning aan van 1,5 volt en verdraai de versterkingsregeling tot er op het scherm staat: De spanning is 1,5 volt.

Voor het afregelen kan elke spanning die binnen het bereik valt, worden gebruikt. De potentiometer wordt verdraaid tot de desbetreffende waarde op het scherm verschijnt. Hoewel dit programma werkt, is er met het door Sinclair gebruikte BASIC een betere opzet mogelijk.

De getallenrij kan worden vastgelegd als DATA.

```

10 REM Voltmeter met volle schaal 1,6 volt
20 DIM v(16): LET v=0: LET p=IN 31
30 LET vmax=1.6
40 FOR i=1 to 16: READ v(i)
50 IF p=v(i) THEN LET v=i
60 PRINT AT 0,0;vmax*v/16;" VOLT"
70 NEXT i
80 DATA 119,183,215,231,235,219,187,123,126,190,222,
      238,237,221,189,125
90 RESTORE
100 GOTO 10


```

Het programma vergelijkt de ingelezen waarde met alle getallen die als DATA zijn opgeslagen.

Dit programma is het fundament voor alle verdere voltmeterprogramma's. Afhankelijk van het meetbereik wordt voor andere toepassingen alleen regel 30 aangepast.

26 Windsnelheidsmeter

De snelheid van de wind is een grootte die niet eenvoudig te meten is. Een windsnelheidsmeter vraagt nogal wat mechanisch geknutsel. Toch is het bouwen hiervan niet echt moeilijk. Het hier omschreven ontwerp is gebaseerd op een aandrijfmotortje uit een cassette recorder. Deze motortjes zijn in de dump-handel ruim voorradig, getuige de advertenties in elektronicabladen. Het motortje van een cassette recorder draait erg licht en leent zich bij uitstek als dynamo. Het draaien van de as levert een prachtige gelijkspanning op. Een experimentje laat dat zien. Sluit een LED aan op de motordraden. Draai aan het asje en de LED licht op. Niet...? Draai dan eens de andere kant uit. Draai niet te fanatiek want dan straalt de LED zijn laatste licht uit. Sluit in plaats van de LED eens een ander recordermotortje aan. Het draait mee als de as van het eerste motortje wordt ge-



Afbeelding 26.1 Eén ping-pongballetje, een motortje en wat hulpmateriaal zijn voldoende voor deze windkrachtmeter.

draaid. Daardoor levert het stroom aan de eerste motor die op zijn beurt blijft draaien en stroom gaat leveren. Na een klein zetje houden de motortjes elkaar in beweging tot ze versleten zijn. Of niet soms?

Terug naar de wind. Op onze windsnelheidsmeter zit een rotor die uit twee halve ping-pongballetjes bestaat (afb. 26.1) die op een staafje zijn gemonteerd. Om regenwater geen toegang tot de lagers te verschaffen, steken we het staafje door een kapje (bijvoorbeeld de dop van een spuitbus) dat over de motor valt. De meter is klaar. De motordraden worden, na verlenging, op de A/D-ingang aangesloten. Afhankelijk van de draairichting van de as voert één van de draden een positieve spanning. Deze draad wordt met de ingang verbonden. De andere draad komt aan de massa-aansluiting.

Een windmeterprogramma dat de windsnelheid in een histogram uitzet, kan erg mooi worden gemaakt.

```
10 REM windsnelheidsmeter
20 BORDER 5
30 DIM v(16)
40 LET vmax=100: LET v=0: LET j=6
50 LET p=IN 31

60 PLOT 0,0
70 FOR n=1 to 16
80 DRAW 0,10: DRAW 5,0: DRAW -5,0
90 NEXT n
100 PLOT 0,0: DRAW 255,0

110 FOR i=1 TO 16
120 READ v(i)
130 IF p=v(i) THEN LET v=i
140 PRINT AT 0,0; INK 2; "WINDSNELHEID ";
    INT(vmax*v/16); " m/s"
150 IF j=255 THEN GOTO 230
160 NEXT i
170 PLOT j,0: DRAW INK 4; 0,vmax*v/16
180 LET j=j+1
190
200 DATA 119,183,215,231,235,219,187,123,126,190,222,
    238,237, 221,189,125
210 RESTORE
220 GOTO 40
230 CLS: LET j=6: RESTORE : GOTO 40
```

Het eerste blokje legt een aantal waarden vast, terwijl het tweede blok een assenkruis tekent.

Het gedeelte vanaf regel 110 vergelijkt de in regel 50 gemeten p met de onder DATA opgeslagen getallen en stelt de bijbehorende windsnelheid vast. Regel 170 maakt er een kolommetje van.

Regel 190 is opzettelijk open gelaten. Door hier een PAUSE-opdracht te plaatsen, kan het tijdsinterval tussen de metingen worden vastgelegd.

```
190 PAUSE 3000
```

Elke minuut wordt nu een meting verricht. Op het scherm is dan ruimte voor een histogram dat ongeveer vier uur samenvat.

Het is mogelijk om elk gevuld beeldscherm weg te schrijven naar de cassette-recorder met

```
SAVE"plaatje"SCREEN$
```

Ophalen van de prent kan met

```
LOAD"plaatje"SCREEN$
```

Dit moet u echt eens proberen. Het is een prachtig gezicht hoe het beeld wordt opgebouwd volgens het patroon dat we in een eerder hoofdstuk al hebben beschreven.

De echte weerkundigen kunnen door de beelden op te slaan een uitgebreide databank van windgegevens opbouwen. De computer kan elk gevuld beeldscherm via de COPY-opdracht ook op de printer zetten. Wellicht bent u de nieuwe Pelleboer in spe.

27 Aquariumregeling

Nu de Spectrum is uitgerust met een analoge ingang, kunnen we allerlei processen automatiseren. Omdat niet iedereen een petrochemische fabriek tot zijn beschikking zal hebben om oefeningetjes mee te doen, is als voorbeeld een tropisch aquarium gekozen.

In zo'n aquarium worden hoge eisen gesteld aan de temperatuur. Deze wordt voortdurend gemeten en bijgeregeld. Een defecte verwarming kan een ramp veroorzaken zodat een reserve-element, dat in geval van nood wordt bijgeschakeld, geen luxe is. Er is een beluchtingspomp die vanwege het lawaai 's nachts misschien moet worden uitgeschakeld. En dan is er natuurlijk nog de verlichting die bij voorkeur natuurgetrouw langzaam aan en langzaam uit moet gaan. Detectie van de waterstand kan het geheel completeren.

Natuurlijk is er een uitgebreid monitor- en signaleringssysteem aanwezig om de gang van zaken in het aquarium bij te houden. Al gauw zal blijken dat het kleurenscherm boeiender is dan het aquarium zelf, zodat op den duur een enorme kostenbesparing kan worden gerealiseerd vanwege het stoppen van de natte hobby.

Een waarschuwing is op z'n plaats. De Spectrum met interface is een ideale combinatie als procescomputer. Een grote onzekerheid wordt echter gevormd door de netspanning. Kleine netstoringen kunnen al catastrofaal zijn. Voor serieuze toepassingen is een standby-voeding met accu's onontbeerlijk.

Een programma dat de gang van zaken bijhoudt en regelt, kan uitgebreid grafisch worden ondersteund. Een natuurgetrouwe weergave van het onderwatertafereel gaat misschien iets te ver. Een suggestieve afbeelding waarin alle variabelen duidelijk zijn af te lezen is wél erg aantrekkelijk. Vooral de temperatuursignalering kan met kleuren worden benadrukt. Het volgende programma moet als een handreiking worden gezien. Hoewel de meeste aquariumfuncties opgenomen zijn, zullen persoonlijke wensen in een door uzelf te schrijven programma veel beter tot hun recht komen.

```
10 REM aquarium
20 DIM i$(704)
30 PRINT AT 0,0; OVER 1; PAPER 1; i$
40 PLOT INK 7; 0,175: DRAW INK 7; 0,-175: DRAW INK
    7; 255,0: DRAW INK 7; 0,175
```

```

50 BORDER 1
60 FOR i=1 TO 5
70 PRINT AT 20-2*i,1; PAPER 1; INK 7; 19+i; "[ ][]"
80 NEXT i
90 PRINT AT 8,1; PAPER 1; INK 6; "TEMPERATUUR"
100 FOR i=1 TO 3
110 PRINT AT 19-i,8; INK 2; "■"
120 NEXT i
130 FOR i=4 TO 6
140 PRINT AT 19-i,8; INK 4; "■"
150 NEXT i
160 FOR i=7 TO 9
170 PRINT AT 19-i,8; INK 2; "■"
180 NEXT i
190 PRINT AT 6,15; INK 4; "■ ■ ■ ■ ■ ■ ■ ■ ■ ■
    ■ ■ ■ ■ ■ ■"
200 PRINT AT 5,15; PAPER 1; INK 6; "WATERNIVEAU"
210 PRINT AT 1,1; INK 7; "■ ■ ■ ■ ■ ■ ■ ■ ■ ■
    ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■ ■
    ■"
220 PRINT AT 1,13; "LICHT"
230 POKE 23672,0: POKE 23673,0: POKE 23674,0
240 LET s=0: LET m=0: LET h=0: LET q=0
250 DEF FN m(x,y)=(x+y+ABS (x-y))/2
260 DEF FN u()=(65536*PEEK 23674+256*PEEK 23673+PEEK
    23672)/50
270 DEF FN t()=FN m(FN u(),FN u())
280 LET t=INT FN t()
290 REM minuten
300 LET s=t
310 IF s>60 THEN LET s=s-60
320 IF s>60 THEN GOTO 310
330 IF s=60 THEN LET s=0
340 IF s<>q AND s=0 THEN LET m=m+1
350 REM uren
360 LET q=s
370 IF m=60 THEN LET h=h+1
380 IF m=60 THEN LET m=0
390 IF h=24 THEN LET h=0
400 PRINT AT 16,20; PAPER 1; INK 6; "TIJD"
410 PRINT AT 18,20;h;"-";m;"-";s;" "

```

```

420 DIM v(16)
430 LET p=IN 31
440 IF p=255 THEN PRINT AT 21,4; FLASH 1; INK 7;
      PAPER 2; "CONTROLEER INTERFACE !!!"
450 FOR i=1 TO 16
460 READ v(i)
470 IF p=v(i) THEN LET v=i
480 NEXT i
490 FOR i=1 TO 3
500 PLOT 47+i,24: DRAW INK 0;0,72
510 NEXT i
520 PLOT 48,24: DRAW INK 5;0,4*v
530 PLOT 49,24: DRAW INK 5;0,4*v
540 PLOT 50,24: DRAW INK 5;0,4*v
550 PLOT 0,0: DRAW PAPER 1; INK 7;255,0
560 IF v<5 OR v>10 THEN GOSUB 1000
570 DATA 119,183,215,231,235,219,187,123,126,
      190,222,238,237,221,189,125
580 RESTORE
590 GOTO 250

1000 REM alarm
1010 PRINT AT 21,4; FLASH 1; PAPER 2; INK 7;
      "CONTROLEER AQUARIUM !!!"
1020 FOR i=1 TO 100
1030 BEEP .01,i/2
1040 NEXT i
1050 RETURN

```

Het eerste blok vult de regels tot en met 220. In dit programmadeel wordt het aquarium met de verschillende functies getekend. Aangegeven zijn het water-niveau, de temperatuur, het licht en de tijd. Het tweede blok verzorgt de tijdmeting en houdt een digitaal uurwerk op het scherm bij. Deze tijdfunctie kan worden benut bij een programma-uitbreiding die het licht aan- en uitschakelt. Ook een voedermechanisme kan vanuit de tijdfunctie worden bestuurd.

Het programmadeel dat op regel 420 begint, kijkt naar de analoge ingang. Het gemeten getal p wordt vergeleken met de in het programma aanwezige DATA (regel 570). Hiertoe wordt zestien maal een DATA-element opgehaald. Zodra er een getal wordt opgehaald dat gelijk is aan de gemeten waarde, krijgt de variabele v de waarde van de telgrootte i.

Het getal v wordt gebruikt om de hoogte van de kwikdraad vast te stellen. De kwikdraad wordt gePLOT in de regels vanaf 520 nadat de oude aflezing in regel 500 is verwijderd. Om een duidelijke temperatuurindicatie te krijgen, is de kwikdraad drie pixels breed gemaakt.


We gaan even terug naar regel 100. Het onderdeel dat hier begint, tekent een verticale rode balk met in het midden een groen gedeelte. Deze balk staat naast de kwikdraad en geeft aan binnen welke grenzen de temperatuur moet blijven. In regel 560 gebeurt de eigenlijke detectie. Als de in te stellen temperatuurgrenzen moeten worden veranderd, moet dat dus op twee plaatsen in het programma gebeuren. Wordt een te hoge of te lage temperatuur gemeten, dan komt de subroutine vanaf regel 1000 in werking. Een rood knipperende waarschuwing komt in beeld die akoestisch wordt ondersteund. Nu kan het gebeuren dat de storing spontaan wordt opgeheven. In dat geval blijft er toch een, zij het wat minder opvallende, waarschuwing in beeld. In dat geval kan bij thuiskomst de installatie alsnog grondig worden nagekeken.

De afregeling

Om te beginnen, wordt de versterkingsregelaar helemaal teruggedraaid (linksom). De versterking is dan 1x. Een temperatuurgevoelige weerstand (NTC), met een weerstandswaarde bij kamertemperatuur van 4k7, wordt op de in het vorige hoofdstuk aangegeven wijze aangesloten. Start nu het aquariumprogramma. De instelbare weerstandsdeler (de verticaal gemonteerde instelpotentiometer) wordt nu verdraaid tot de kwikdraad op het scherm verschijnt. We zetten hem op de temperatuur die op dat ogenblik heerst. Draai nu de versterkingsregelaar een aantal slagen rechtsom. De kwikdraad zal hierdoor stijgen. Deze stijging wordt weer weggewerkt met de verticaal gemonteerde weerstandsdeler. Als de NTC nu wordt verwarmd door hem vast te pakken, zal de thermometer op het scherm een stijging laten zien. Voor het precies afregelen hebben we twee bekende temperaturen binnen het meetbereik nodig (twee bakjes water).

De beschreven procedure wordt een aantal malen herhaald, waarbij met de verticale potentiometer de onderwaarde wordt ingesteld, om vervolgens met de horizontale potentiometer de hoogste temperatuur in te stellen.

Hoe zit het nu met de signalering van de waterstand? Een transistor over de tem-



Afbeelding 27.1 Signalering van de waterstand.

peratuurgevoelige weerstand kan daar een oplossing voor geven, zoals afbeelding 27.1 toont.

Als het waterniveau beneden een bepaalde waarde zakt, zal de transistor worden opengestuurd omdat er een basisstroom gaat lopen. De analoge ingang zal het effect van de opengestuurde transistor uitleggen als een flinke temperatuurstijging. Resultaat: Het alarm komt in.

Een waarschuwing is echter op zijn plaats. Vissen zijn gevoelig voor elektrische velden. Een directe stroomgang door het water, al is dit maar een stuurstroompje, is daarom af te raden. Een vlottertje met ingebouwde schakelaar geeft dan ook een betere oplossing. Het maakcontact kan direct over de NTC worden geschakeld (een breekcontact moet in serie met de NTC). De transistor kan dan ook vervallen.

De hier geschetste aquariumregeling geeft slechts een globaal beeld van de mogelijkheden met de interface met uitbreiding. Zo is de regeling van het licht nog niet aan de orde geweest. In het volgende hoofdstuk zullen we een uitbreiding behandelen waarmee twee lampengroepen vloeiend en naar behoefte zeer langzaam kunnen worden aan- en uitgeschakeld.


28 Programmeerbare diavloeiër

De reeds omschreven lichtnet-interface stelt ons in staat om lampengroepen door de Spectrum te laten schakelen. Het schakelen geschiedt hierbij echter abrupt. Voor veel toepassingen is het wenselijk een lichtniveau in te kunnen stellen. Ook kan een vloeiende aan/uit-overgang gewenst zijn. Aquariumverlichting is hier een voorbeeld van. Maar ook de diashow wordt opmerkelijk verlevendigd met vloeiende lichtovergangen en andere lichteffecten.

We denken hierbij aan een opstelling van twee diaprojectoren en een cassette-recorder. De projectoren worden door de computer gestuurd. Het programma wordt hierbij gesynchroniseerd door de bijbehorende geluidsband.

Met in elkaar overvloeiende beelden, kunnen bewegingen suggestief worden weergegeven met een soort slow motion effect. Een bijzonder indrukwekkend effect wordt verkregen door op een geprojecteerde afbeelding van een dreigende onweerslucht, met de andere projector een flitsende bliksemschicht te projecteren. Het bliksemeffect wordt verkregen door de lamp enkele keren kort in te schakelen. Een passende geluidsondersteuning garandeert een bloedstollend effect.

Een voordeel van de computersturing is nog dat de lamp in de uit-stand niet volledig gedoofd hoeft te worden. Dat komt de levensduur ten goede en maakt het mogelijk om flitseffecten te gebruiken zonder dat dit nadelig is voor de lamp.



Afbeelding 28.1 Twee maal vier uitgangen.

De schakeling

Op de interfaceprint is een analoog uitgangssignaal beschikbaar. Voor ons doel hebben we echter twee analoge spanningen nodig die onafhankelijk van elkaar kunnen worden geregeld.

Met de kennis die we van R-2R netwerken hebben, is dat niet moeilijk te verwezenlijken.

We verdelen de acht uitgangslijnen in twee groepen van vier en voorzien beide groepen van een eigen weerstandsnetwerk. Met vier uitgangen zijn zestien combinaties mogelijk, zie afbeelding 28.1.

Lijn					Lijn				
0	1	2	3	Decimaal	4	5	6	7	Decimaal
0	0	0	0	0	0	0	0	0	0 = 0*16
1	0	0	0	1	1	0	0	0	16 = 1*16
0	1	0	0	2	0	1	0	0	32 = 2*16
1	1	0	0	3	1	1	0	0	48 = 3*16
0	0	1	0	4	0	0	1	0	64 = 4*16
1	0	1	0	5	1	0	1	0	80 = 5*16
0	1	1	0	6	0	1	1	0	96 = 6*16
1	1	1	0	7	1	1	1	0	112 = 7*16
0	0	0	1	8	0	0	0	1	128 = 8*16
1	0	0	1	9	1	0	0	1	144 = 9*16
0	1	0	1	10	0	1	0	1	160 = 10*16
1	1	0	1	11	1	1	0	1	176 = 11*16
0	0	1	1	12	0	0	1	1	192 = 12*16
1	0	1	1	13	1	0	1	1	208 = 13*16
0	1	1	1	14	0	1	1	1	224 = 14*16
1	1	1	1	15	1	1	1	1	240 = 15*16

Het onderscheiden van de twee kanalen blijkt opmerkelijk eenvoudig. Voor projector 1 gebruiken we de opdracht OUT 31,n waarbij n varieert van 0 tot en met 15.

Voor de tweede projector ziet de opdracht er ongeveer hetzelfde uit: OUT 31,m*16 waarin m ook van 0 tot en met 15 kan variëren. Er zijn dus voor elke projector zestien discrete helderheidsstappen. Voor een combinatie van twee projectoren worden de getallen opgeteld, dus OUT 31,n+16*m.

Bijvoorbeeld:


```

1Ø FOR n=Ø TO 15
2Ø m=15-n
3Ø OUT 31,n+16*m
4Ø NEXT n

```


Het getal 15 betekent dan dat de lamp uit is. Zoals de tabel laat zien, zijn de kanalen in dat geval allemaal hoog. Deze toestand gebruiken we voor het dia-transport.


De elektronica die nodig is om de stuurspanning om te zetten in lichtintensiteit is in afbeelding 28.2 weergegeven.



Afbeelding 28.2 Het schema voor het verkrijgen van de stuurspanning.

Omdat voor elke nuldoorgang van de netspanning de gatestroom van de triac wordt onderbroken, zal de triac op elke nuldoorgang afschakelen (afb. 28.3). Echter, de R-C-tijd is zodanig gekozen dat tijdens de halve periode van de netspanning de 555-uitgang omklapt, waardoor de triac wordt ontstoken. Met de instelpotentiometer wordt deze tijd zo ingesteld dat de lamp juist volledig brandt, dat wil zeggen dat de 555-pulstijd zeer kort is. Bij deze afregeling moet de sturingang (aansluiting 5) aan massa liggen. Hiermee zijn we dan tevens bij het belangrijkste deel van de schakeling aangeland. De sturingang wordt verbonden met de regelspanning die met het R-2R-netwerk wordt opgewekt. Door de

stuurspanning te verhogen, wordt ook de 555-pulslenkte vergroot. De triac blijft daardoor langer uit en de computergestuurde dimmer is een feit, zie afbeelding 28.4.




100 Hz
R-C tijd 555
geleiding TRIAC


Afbeelding 28.3 Het schakelen van een triac.

Rest nog de diawisseling. Zoals afgesproken dient de wisseling plaats te vinden als alle interfacelijnen hoog zijn. De stuurspanning is dan maximaal en de lichtsterkte minimaal. Een viervoudige NAND-poort is bij uitstek geschikt voor dit doel.

Slechts als alle ingangen hoog zijn, gaat de uitgang omlaag. Een transistortrapje stuurt het relais, dat op zijn beurt het wisselmechaniek in de projector in werking stelt.



Afbeelding 28.4 De computergestuurde dimmer.



Afbeelding 28.6 De ingreep in de diaprojector.

```

80 FOR n=0 TO 14
90 OUT 31,n+16*14
100 PAUSE 25
110 NEXT n

```

In regel 10 wordt het diawisselmechanisme in werking gesteld en door regel 20 vastgehouden gedurende eentiende seconde. Als deze tijd langer wordt gemaakt, vindt er bij veel projectoren transport achteruit plaats. Handig om te weten. In het blokje van regel 30 t/m 60 wordt n steeds met 1 verlaagd. De lamp gaat feller branden tot hij bij 0 op volle sterkte brandt. De lengte van dit proces wordt geregeld door regel 50. PAUSE 25 staat voor een halve seconde. De vijftien stappen vragen totaal dan ook 7,5 seconde. Let erop dat de toevoeging $16*14$ in de OUT-opdracht ervoor zorgt dat de tweede projector gedoofd blijft.

Het derde blok laat de tweede projector even oplichten. Het spreekt voor zich dat hier de toevoeging $16*0$ evengoed weggelaten kan worden.


In het laatste blok wordt de lamp langzaam gedimd. Let erop dat we dimmen t/m 14. Immers bij 15 vindt er weer diatransport plaats en we hadden maar één dia.

Waarschuwing!

De hier beschreven schakeling werkt volkomen veilig in combinatie met moderne laagspanningsprojectoren.

Oudere projectoren zijn soms uitgerust met een 220 volt-lamp. Hoewel de schakeling ook hier normaal zal werken, mag de koppeling met de computer in dit geval **nooit** rechtstreeks plaatsvinden. Dit geldt ook voor die toepassingen waarbij 220 volt-lampen, verwarmingselementen of motoren worden aangesloten. Zie hiervoor de opmerkingen in hoofdstuk 21.

De meest eenvoudige ontkoppeling is aan te brengen als opsteekprintje op de interfaceprint. Op dit printje worden dan vier optocouplers gemonteerd die voor de nodige isolatie zorgdragen. Laat u in géén geval verleiden zonder afdoende isolatie met de lichtnetspanning te spelen.



Afbeelding 28.7 De gemonteerde print.

29 Uit is in: Een lichteffect voor de disco

In de voorafgaande hoofdstukken zijn de schakelingen als afzonderlijke bouwstenen toegepast. In dit hoofdstuk zullen we de computer gebruiken om aan de hand van een binnenkomend signaal de uitgang te sturen. Uiteraard na diverse bewerkingen door een programma. De titel 'uit is in' slaat dan ook niet alleen op een avondje stappen, hij beschrijft ook de werking van een lichteffect, speciaal voor de disco.



Doordat de analoge ingangsschakeling met een gevoelige voorversterker is uitgerust, kunnen we er een microfoon op aansluiten. Maar ook het signaal afkomstig van een cassette recorder of versterker kan als stuursignaal worden gebruikt. Aan de uitgang zouden we de viervoudige lichtnetinterface kunnen gebruiken. Het effect is dan flitsend. Wel aardig, maar niet wat we zoeken. Door de tweekanaals-diavloeiër met de uitgang te verbinden, zijn levendiger effecten te realiseren. Eenvoudige, goedkope diaprojectoren zijn bovendien prachtige lichtbronnen waarmee speciale effecten bereikt kunnen worden. Bij gebruik van 220 volt-spotjes sluiten we de diavloeiër uiteraard aan via een optische ontkoppeling.

Door zonder meer een audiosignaal op de analoge ingang aan te sluiten, overvoeren we de computer met een chaos van wisselende spanningen. Liever hebben we een signaal dat de geluidsstrekte representeert. Op de analoge ingangsprint is een en ander eenvoudig te realiseren. We gaan ervan uit dat de schakeling min of meer permanent voor het gestelde doel gebruikt zal worden. Een kleine ingreep op de print zal dan ook geaccepteerd moeten worden.

Afbeelding 29.1 laat de wijziging zien.

De verbinding tussen de versterkeruitgang en de sturingang van de UAA 170 wordt verbroken. Een diode zorgt voor enkelvoudige gelijkrichting. Met dit gelijkgerichte signaal wordt een condensatortje opgeladen. Het zal duidelijk zijn dat de condensator zeer snel de piekwaarde van het signaal aanneemt. We moeten dan ook zelf voor ontlading zorgen. De weerstand parallel aan de condensator knapt dit karweitje op. De onderdeeljes kunnen in de buurt van de sturingang op de print een plaatsje vinden. De keuze van de componentenwaarden beïnvloedt de snelheid waarmee het ingangssignaal wordt gevolgd. Eventueel kan de weerstand als instelpotentiometertje worden uitgevoerd.

Een kristal microfoon (of een andere signaalbron) kan rechtstreeks op de verster-



Afbeelding 29.1 De swingende wijziging voor discogebruik.

keringang worden aangesloten. De extra ingangsdeler wordt dus niet gebruikt. De diavloeiërs worden zonder wijzigingen aangesloten. Om de juiste instelling van de versterkingsregelaar te bepalen, is een klein afregelprogramma afgedrukt.

```

10 REM afregeling versterking
20 LET v=0
30 DIM v(16)
40 LET p=IN 31
50 FOR i=1 TO 16
60 READ v(i)
70 IF p=v(i) THEN LET v=i
80 PRINT AT 0,0;v;" "
90 NEXT i
100 DATA 119,183,215,231,235,219,187,123,126,190,
        222,238,237,221,189,125
110 RESTORE
120 GOTO 40

```

We regelen de gevoeligheid zo af dat het getal op het scherm tussen 15 en 16 twijfelt. Wordt er nu in de microfoon geblazen dan verschijnt er een lager getal. Hoe harder het geluid, hoe lager het getal.

Door het ingangssignaal in een programma te bewerken, kunnen allerlei bizarre effecten worden teweeggebracht. Het programma geeft daar een voorbeeld van. De twee uitgangen worden in tegenfase uitgestuurd. Door de lichtbronnen van de afzonderlijke kanalen op enige afstand van elkaar te monteren en op een bepaalde plaats te richten, ontstaat een aardig effect. Het licht wisselt op het ritme van de muziek. U begrijpt dat het gebruik van twee elkaar aanvullende kleuren zoals rood en groen het geheel nog dynamischer maakt. Bij het gebruik van diaprojectoren kan de muziek zelf zorgen voor een wisselend kleurenbeeld door het diawisselmechanisme bij het programma te betrekken. Als er niet wordt scherp-gesteld, ontstaan er kleurenvlekken in het ritme van de muziek. Bij scherpe projectie kan een wand of vloerdeel als levendig decor worden gebruikt. Wat dacht u van flitslicht dat door de diawisselrelais wordt bediend?

Bij goed gebruik moet het mogelijk zijn de dansende horden aan de rand van een zenuwinzinking te brengen. Mocht dat niet lukken, probeer de schakeling dan eens (zonder flitslicht) bij wat rustiger muziek. Bij gebruik van rood en groen, dicht bij elkaar opgesteld, ontstaat een prachtige verstemming van de kleur van het licht.

Voor de belichting van een zogeheten 'swinging poster' is de geschetste opstelling natuurlijk ook uitstekend geschikt. (Voor diegenen die niet elke dag in een disco komen: Een swingende poster heeft twee dezelfde ten opzichte van elkaar enigszins verschoven afbeeldingen in de kleuren rood en groen. Door de prent afwisselend met rood en groen te belichten, is het alsof de figuur van de ene positie naar de andere springt. Gaat het wisselen op het ritme van de muziek, dan lijkt het of de poster de pan uit swingt.)

Het programmavoorbeeld zorgt ervoor dat de twee lampengroepen, bij afwezigheid van signaal, in een rustig tempo wisselen. Een onopvallend sfeereffect. Zodra echter het geluid de ruimte vult, komt de computer in extase.

```
10 REM disco
20 LET v=0
30 DIM v(16)
40 FOR s=1 TO 180 STEP 5
50 LET p=IN 31
60 FOR i=1 TO 16
70 READ v(i)
80 IF p=v(i) THEN LET v=i
90 PRINT AT 0,0;v;" "
100 NEXT i
110 DATA 119,183,215,231,235,219,187,123,126,190,222,
      238,237,221,189,125
```



```

120 RESTORE
130 IF v<16 THEN GOSUB 8000
140 LET f=INT(14*SIN(s*2*PI/360))
150 OUT 31,f+16*(14-f)
160 NEXT s
170 GOTO 40
8000 OUT 31,(v-1)+16*(15-v)
8010 RETURN

```

Omdat het programma geen nieuwe elementen bevat, kan de uitleg beperkt zijn. Het uitlees- en decodeergedeelte voor de analoge ingang zijn we al diverse malen tegengekomen. Als er geen signaal is, zorgt de ingebouwde sinusfunctie voor het sfeerlicht waarbij de twee lampen(groepen) afwisselend in een rustig tempo oplichten. Bij gebruik van rood en groen ontstaat er een voortdurend veranderende kleur. Zodra er echter signaal aangeboden wordt, zet de subroutine dit om in een snel volgende lichtintensiteit. Evenals bij de diavloeiër is bij deze schakeling geen beeldscherm nodig. De TV hoeft niet aangesloten te zijn. In het programma is dan ook niets aan grafische ondersteuning gedaan. Natuurlijk kan de beeldbuis wel degelijk bij het complexe lichtspel betrokken worden. Het maken van leuke grafische effecten laat ik graag aan u over. (Denk eens aan de truc waarmee de schermkleur plotseling veranderd kan worden.)

Een rechtgeaarde discjockey met wat computergevoel zal in het hier beschreven programma slechts een summiere handleiding zien. Ook de sologitarist zal zo zijn eigen geluidsondersteuning wensen. Als de basgitarist het computerorgel voor zich opeist, kan hij het hol dreunende gevoel in de omgeving van het middenrif prachtig visueel ondersteunen. Maak er wat moois van.


30 Verlengprint

Een handig hulpmiddel bij het experimenteren met de verschillende signalen op de Spectrum-uitgang is de verlengprint.


Een stukje dubbelzijdig printplaat met kopersporen op de juiste afstand wordt aan één zijde van een Spectrum-connector voorzien. De print wordt met plakplastic bedekt waarbij er versprongen openingen worden aangebracht. In deze openingen kan, zonder gevaar op sluiting tussen de sporen, een meetpen worden geprikt. Ook kan ter plaatse van zo'n opening een draadje worden gesoldeerd dat met een experimentele schakeling kan worden verbonden. Uitbreidingen kunnen normaal worden aangesloten zodat ook het dataverkeer met uitbreidingen kan worden bestudeerd. Met het ongemak dat de helft van de aansluitingen zich aan de onderzijde bevindt, zult u moeten leren leven (tenzij u er iets op vindt).

Tot slot


Als u aan dit boek even veel plezier beleeft als de schrijver zelf, heeft u de gave om op het brede werkteerrein van computers iets zinvols tot stand te brengen. De combinatie computer en elektronica biedt fascinerende mogelijkheden. Hoewel met de hier beschreven schakelingen vele praktische zaken kunnen worden gerealiseerd, zijn er nog veel meer gevallen waarin deze schakelingen géén oplossing bieden. In dat geval zult u zelf aan de slag moeten. Door een probleem goed te omschrijven zijn er vaak verrassend eenvoudige oplossingen te bedenken. Zoals in de inleiding al werd gesteld, kan ondersteuning in club-verband een stimulant vormen om ook ingewikkelde interfaceproblemen aan te pakken. Vergeet hierbij nooit dat een uitgewerkte oplossing slechts beperkte waarde heeft als hij niet vergezeld gaat van een goede documentatie. Schrijf het op, zodat ook anderen er plezier aan kunnen beleven. Wellicht kan een volgend boekwerk dan van uw hand zijn.




Printontwerp dimmer



Printontwerp A/D-omzetter




Printontwerp lichtopnemer muis




Printontwerp lichtnetschakelaar

- * Het complete schema van de interface die in hoofdstuk 4 staat beschreven is niet in het boek opgenomen. De diverse onderdelen van de interface staan in de voorgaande hoofdstukken beschreven.
- * Hieronder staat het bestukkingsschema van de interface-print die bij het boek is geleverd. De opbouw van de print staat in hoofdstuk 4 beschreven.





De zwarte bolletjes zijn printpennen (24 stuks), de dikke zwarte lijnen zijn draadbruggen.

In hoofdstuk 5 staat een uitgangsmontitor beschreven. De printlayout hiervan is in het boek weggevalen. De printlayout en de componentenopstelling van de uitgangsmontitor:



- * Hieronder staat het schema van de 4093 die in de hoofdstukken 15 en 16 ter sprake komt:





De print-lay-out van het stekkerprintje waarmee de verbinding tussen de flat-cable en de interface-print kan worden gemaakt, staat rechtsbovenaan op bladzijde 105 van het boek.

De ZX Spectrum is een uitermate veelzijdige machine met mogelijkheden die in veel gevallen slechts door de gebruiker zelf worden begrensd. Dit boek behandelt een aantal elektronica-projecten waarmee de gebruiksmogelijkheden van de Spectrum met een nieuwe dimensie worden uitgebreid. Gebaseerd op een universele interface worden diverse meet- en regelschakelingen behandeld. De toepassingen strekken zich uit op vele gebieden. Van lichteffecten voor toneel en disco tot een veelzijdige aquariumregeling. Van joy-stick en lichtpen als programmahulp tot toepassingen bij het meten en regelen van natuurkundige processen. Vele programma's illustreren het gebruik van de behandelde schakelingen. Uiteraard wordt hierbij dankbaar gebruik gemaakt van de grafische mogelijkheden van de Spectrum. De ontwerpen zijn voorzien van een print-layout waarmee op snelle en doeltreffende wijze de vervaardiging van prints ook voor de beginner op elektronica-gebied geen probleem hoeft te vormen.

De ZX Spectrum is een uitermate veelzijdige machine met mogelijkheden die in veel gevallen slechts door de gebruiker zelf worden begrensd. Dit boek behandelt een aantal elektronica-projecten waarmee de gebruiksmogelijkheden van de Spectrum met een nieuwe dimensie worden uitgebreid. Gebaseerd op een universele interface worden diverse meet- en regelschakelingen behandeld. De toepassingen strekken zich uit op vele gebieden. Van lichteffecten voor toneel en disco tot een veelzijdige aquariumregeling. Van joy-stick en lichtpen als programmahulp tot toepassingen bij het meten en regelen van natuurkundige processen. Vele programma's illustreren het gebruik van de behandelde schakelingen. Uiteraard wordt hierbij dankbaar gebruik gemaakt van de grafische mogelijkheden van de Spectrum. De ontwerpen zijn voorzien van een print-layout waarmee op snelle en doeltreffende wijze de vervaardiging van prints ook voor de beginner op elektronica-gebied geen probleem hoeft te vormen.

ISBN 90 201 1874 9


W. H. M. van Dreumel

Elektronica-projecten voor de ZX SPECTRUM (+)



Elektronica-projecten voor de ZX SPECTRUM en ZX SPECTRUM+

W.H.M. van Dreumel




KLUWER TECHNISCHE BOEKEN

Elektronica projecten voor de **ZX SPECTRUM** en **ZX SPECTRUM+**

W.H.M. van Dreumel

met
interface
print



KLUWER TECHNISCHE BOEKEN

**Elektronicaprojecten
voor de
ZX SPECTRUM
en ZX SPECTRUM (+)**